

UNITED STATES DEPARTMENT OF THE INTERIOR
GEOLOGICAL SURVEY

BASIC Programs to Calculate Gravity and Magnetic Anomalies
for 2 1/2 - dimensional Prismatic Bodies

by

David L. Campbell

Open-File Report 83-154

1983

This report is preliminary and has not been reviewed for conformity with U.S. Geological Survey editorial standards and nomenclature.

Use of brand names and model numbers in this report is for the sake of description only, and does not constitute endorsement by the U.S. Geological Survey.

Although the programs listed here have been extensively tested, the Geological Survey makes no guarantee whatsoever of correct results.

BASIC Programs to Calculate Gravity and Magnetic Anomalies

for 2 1/2-dimensional Prismatic Bodies

by David L. Campbell

SUMMARY

This report lists six Talwani-type programs and test cases for forward calculation of gravity and total-field magnetic anomalies due to horizontal prismatic bodies. These programs were written in BASIC language for use with the Hewlett-Packard Model 85 computer.¹ The programs all have the same structures, allowing successive modifications of source bodies so as to match observed anomaly curves. Subroutines specific to HP computers are included to sketch the bodies and resulting fields, and to store cases of interest on magnetic tape cassettes. A useful matrix formulation for nonsymmetrical 2 1/2-dimensional magnetic bodies is included in the discussion.

DESCRIPTION

So-called "Talwani-type" computer programs (programs to calculate potential-field anomalies over horizontal prismatic source bodies having planar faces) were introduced by Heirtzler and others (1962) and by Talwani and Heirtzler (1964). These early programs were two-dimensional (2D), in that the source bodies were taken to be infinitely long in the strike direction. Later refinements (Shuey and Pasquale, 1973, Talwani, 1974; Cady, 1979) allowed the source bodies to be truncated in the strike direction by vertical planar ends ("2 1/2 D geometry"). The analytic formulation followed by the 2 1/2 D programs given here is that of Rasmussen and Pedersen (1979).

The programs in Appendix I are written in BASIC language for a Hewlett-Packard Model 85 computer with Printer/Plotter and Input/Output ROMs. Programs 2DGRAV and 2DMAG calculate anomalies for 2D source bodies, programs 2HDGRV and 2HDMAG deal with symmetrical 2 1/2 D bodies (those truncated a finite but symmetrical distance to either side of the profile line), and programs NEWGRV and NEWMAG deal with non-symmetrical 2 1/2-D bodies (those truncated an arbitrary distance from the profile line).

The programs all have the same logical structure as shown in Fig. 1. This structure allows successive modification of the source parameters so the user may attempt to match given anomaly curves. All modifications must be made by the user -- no inversion techniques are used in these programs.

The programs as written allow up to 5 source bodies, each with up to 20 body vertices, and up to 100 field points per profile. These specifications may be changed by changing the appropriate dimensions in the SHORT statements near the beginning of each program. The programs allow either constant altitude or draped surveys. (Typical gravity surveys are considered to be

¹ Manufacturers and model numbers are given for the sake of complete description. This citation does not constitute endorsement by the U.S. Geological Survey.

draped on topography at zero clearance). The effect of source topography may be calculated; under this option, only topography higher than the lowest topographic point is considered.

TAPE AND PLOT SUBROUTINES

Because tape-handling and graphics features tend to be computer-specific, the portions of the programs which handle these functions have been separated into separate subroutines. Users of computers other than the HP85 for which these programs were developed must write and substitute their own subroutines at these locations.

There are two plot subroutines. The first draws a picture of the input geometry in the lower 1/3 of the CRT screen. The profile fills the screen width, and vertical exaggeration is (nearly) 1:1. Thus portions of bodies will not be shown which extend beyond the ends of the profile or to depths greater than (approximately) 1/4 the profile length. The second subroutine plots calculated fields as x's on an x-f(x) graph in the upper 2/3 of the screen. The user must input minimum and maximum field values (FMIN, FMAX) to scale the f(x)-axis. There is one tic mark per field point on the x-axis and a total of ten tic marks on the f(x) axis. The x-axis is placed at zero of the f(x) axis; if FMIN<0, no x-axis is drawn. If the user has input observed field values in response to the prompt near the beginning of the program, they are plotted as a continuous curve on the graph. The program makes no other use of these input observed field values. In order that successful models may be replotted on an auxiliary x-y plotter, with shading and labelling done separately depending on the application, no shading or labelling is done by the plot subroutines.

Subroutines are included to save models of interest on tape, such as those intended for final plots, or intermediate ones produced during a modelling run which is to be resumed later. Tape storage is formatted to allow compatibility between all three gravity programs and all three magnetics programs. Thus the user may approximately model a given case using 2DGRAV, say, which is fastest; then store results on tape and resume modelling via one of the slower programs 2HDGRV or NEWGRV using the tape to input the parameters of the model. (He could also use such a tape as input to 2DMAG, 2HDMAG or NEWMAG to see what magnetic fields his model might generate. In this case, however, he must provide T0, S0, and I0 values to the program during a pause, and also use the "CHANGE SUSC?" option to replace density values stored in array K () with corresponding susceptibility values).

COMMENTS AND WARNINGS

1. The magnetics programs only calculate fields due to induced magnetizations. The effects of permanent magnetizations are not included, nor are those due to demagnetization effects or to effects resulting from interactions between magnetic bodies (see e.g., Sampaio, 1982).

2. The programs sometimes give incorrect answers at field points which happen to lie vertically below a body edge or its extension (this is because the arctangent functions used in the programs are set up to change principal values at the zenith). The error is usually obvious on scanning the profile plot, and can be checked by shifting the questionable field point horizontally by an infinitesimal distance, and re-running the program.

Also, programs NEWGRV and NEWMAG may give wrong answers for field points which lie above a double edge—that is, above a vertical body corner for which either $Y1=0$ or $Y2=0$, exactly (see Fig. 2). In this case a "/ZERO" message is displayed and the answer is usually obviously wrong. Again, the best remedy is to shift the field point infinitesimally away from the objectionable location, and re-run the program.

3. The geometry used by the programs is shown in Fig. 2. Profile direction x is always perpendicular to the strike of the source body (y -direction). Note that z is positive upwards—for example, to represent elevations above sea-level. Inside the program, this is switched to the z -positive-downward convention used in the analytic solutions. As far as the user is concerned, however, x , y , and z form a right-handed coordinate system. In the magnetic programs $S0$ is the angle from magnetic north to strike of the body ($+y$ direction), positive counter-clockwise. Thus the $+x$ direction is into the magnetic northerly half-circle for $0 < S0 < 180^\circ$. In NEWMAG and NEWGRV, the profile is located at $y=0$ and the strike-extends $Y1$, $Y2$ of each body represent coordinates, where $Y1 < Y2$. Thus a profile over the center of 2-km-long body has $Y1=-1$, $Y2=1$. A body with $Y1=Y2$ would have zero length. A body with $Y1=-2$, $Y2=-1$ is located 1-km to the side of the profile in the $-y$ direction.

4. Always input topographic profile points sequentially in the $+x$ direction (from left to right on the CRT screen). If the entire profile range is not covered, the program assumes topography to extend to the left and right at the same level as the nearest topo-point. Up to 20 topo-points may be input, or only 16 if the field due to the topography is to be calculated. To activate the topo-calculation procedure, answer "0" (that is, "no bodies") to the query "HOW MANY BODIES?". The topo-body whose attraction is calculated has a horizontal base at the level of the lowest topo-point, so that the effect of any slabs under this level must be calculated by the user and added separately. In order to suppress possible end-effects, the program adds horizontal slabs one profile-length long to each end of the profile before doing the calculation.

5. In the procedure for changing body corners, the program asks the number of the corner to be changed. At this point you may also add or delete corners. To add corners, be sure to give the next highest corner number for that body. (The program does not check to make sure you counted properly!) To delete a corner, give the negative of that corner's number. To escape to the next body or next segment of the program, give "0".

6. A final, almost trivial, warning—make sure all distances are in the same units, either kilofeet or kilometers. Most "wrong answers" new users found during the program-checking period resulted from entering profile coordinate, X , in km, but elevations or body coordinates, Z , in feet or kilofeet.

NONSYMMETRIC MAGNETIC BODIES

Rasmussen and Pedersen (1979) give equations for non-symmetric magnetic bodies which may be written in the following form:

$$\begin{pmatrix} B_z \\ B_x \\ B_y \end{pmatrix} = \sum_i \begin{pmatrix} C \Delta I_3 & C \Delta I_2 & C \Delta I_4 \\ -S \Delta I_3 & -S \Delta I_2 & S \Delta I_4 \\ C \Delta I_4 & -S \Delta I_4 & (S \Delta I_2 - C \Delta I_3) \end{pmatrix} \begin{pmatrix} J_z \\ J_x \\ J_y \end{pmatrix}$$

Where $\Delta I = I(Y_2) - I(-Y_1)$

$$I_2 = CL - ST$$

and $I_3 = -SL + CT$

In these equations S and C are direction cosines of the i^{th} prism face, L and I_4 are logarithmic functions, T is an arctangent function, \underline{J} is the magnetization vector in the body and \underline{B} the resulting magnetic vector at the field point.

In order to set up an iterative scheme moving from face to face of the prism, it is useful to cast the above equation in the alternative form:

$$\begin{pmatrix} B_z \\ B_x \\ B_y \end{pmatrix} = \sum_i \begin{pmatrix} M_{11} & M_{12} & M_{13} \\ M_{21} & M_{22} & M_{23} \\ M_{31} & M_{32} & M_{33} \end{pmatrix} \begin{pmatrix} L \\ T \\ I_4 \end{pmatrix}$$

On working out the details of this and correcting a sign error in R&P, one finds

$$\begin{pmatrix} -M_{11} \\ M_{12} \end{pmatrix} = \begin{pmatrix} C & S \\ -S & C \end{pmatrix} \begin{pmatrix} J_x \\ J_z \end{pmatrix}$$

$$\begin{pmatrix} M_{21} \\ -M_{22} \end{pmatrix} = \begin{pmatrix} C & S \\ -S & C \end{pmatrix} \begin{pmatrix} J_x \\ J_z \end{pmatrix}$$

Where the matrix $\begin{pmatrix} C & S \\ -S & C \end{pmatrix}$

is also used at an earlier stage to calculate certain auxillary variables (Rasmussen and Petersen's u_1 , u_{1+1} , w_1 and w_{1+1}). In addition

$$\begin{aligned} M_{13} &= CJ_y & M_{32} &= -J_y \\ M_{23} &= -SJ_y & M_{33} &= CJ_z - SJ_x \\ M_{31} &= 0 \end{aligned}$$

These are the equations which are programmed.

REFERENCES

- Rasmussen, R., and Pedersen, L. B., 1979, End corrections in potential field modelling: *Geophys. Prospecting* 27, p. 749-760.
- Shuey, R. T., and Pasquale, A. S., 1973, End corrections in magnetic profile interpretation: *Geophysics* 38, p. 507-512.
- Heirtzler, J. R., Peter, G., Talwani, M., and Zurfeueh, E. G., 1962, Lamont Geol. Obs. (Columbia Univ.) Tech. Report #6, CU-6-62.
- Cady, J. W., 1980, Calculation of gravity and magnetic anomalies of finite-length right polygonal prisms: *Geophysics* 45, v. 10, p. 1507-1512.
- Sampaio, E. E. S., 1982, Magnetic anomalies of two-dimensional bodies in a magnetic half-space: *Geophysics* 47, p. 1229-1234.
- Talwani, M., 1973, Computer usage in the computation of gravity anomalies, in *methods in computational physics*, ed. by B. Alder and others, vol. 13, p. 343-389.
- Talwani, M., and Heirtzler, J. R., 1964, Computation of magnetic anomalies caused by two-dimensional structures of arbitrary shape, in *Computers in mineral industries*, part I, Stanford Univ. Publ. Geol. Sci., vol. 9, p. 464-480.




Fig. 1.--Flow diagram showing the general structure common to all six programs in this report. Steps in ovals are done by subroutines. Steps in broken boxes are optional, and may be bypassed by giving an appropriate response to the computer prompt.




Fig.2.--Sketches illustrating geometry used by the programs. a. the right-handed triad; profile (+X), strike (+Y) and up (+Z). b. plan views showing directions of profile and strike for various strike angles, S_0 . c. plan views showing location of body for Y_1 , Y_2 values used in examples.


```

10 ! PROGRAM 2DGRAV**SEPT,1982*
   **DAVE CAMPBELL, USGS-DENVER
20 CLEAR
30 DISP "2DGRAV FORWARD CALC'N"
40 OPTION BASE 1
50 DEG
60 INTEGER I,J,M,M0,N,N0,N1(5),
   V
70 SHORT X0(20,5),Z0(20,5),X(10
   0),Z(100),F(100),V1(20),V2(2
   0),K(5),Y1(5),Y2(5),H(100)
80 SHORT G,X8,X9,D,V3,W,F0,X1,Z
   1,R1,E,X2,Z2,R2,R6,B,T
90 REAL P1,P2,P,S,C,T2,L
100 SHORT Y8,Y9,Z7,X7,H1,L0,Z8,Z
   9,Z5 ! VARIABLES USED IN SUB
   S
110 IMAGE 2(DDDD.DDD,X),DDDDD.DD
120 G=0 @ H(1)=0
130 DISP "IS DATA ON TAPE? (Y/N)
   "
140 INPUT Y0$
150 IF Y0$="Y" THEN 3160
160 DISP "KFT OR KM?"
170 INPUT Y0$
180 IF Y0$="KFT" THEN G=2.032
190 IF Y0$="KM" THEN G=6.667
200 IF G=0 THEN GOTO 160
210 DISP "***SET-UP FIELD POINTS
   ***"
220 DISP "FIELDX: XMIN,XMAX,DX"
230 INPUT X8,X9,D
240 M0=1+INT((X9-X8)/D)
250 FOR M=1 TO M0
260 X(M)=X8+(M-1)*D
270 NEXT M
280 DISP "INPUT OBS FIELDS? (Y/N
   )"
290 INPUT Y0$
300 IF Y0$#"Y" THEN 350
310 FOR M=1 TO M0
320 DISP "OBS(";X(M);")"
330 INPUT H(M)
340 NEXT M
350 DISP "INPUT TOPO? (Y/N)"
360 INPUT Y0$
370 IF Y0$#"Y" THEN 660
380 ! ***TOPO INPUT**
390 DISP "HOW MANY TOPO POINTS"
400 INPUT V
410 FOR I=1 TO V
420 DISP "TOPO-X(";I;"),TOPO-Z("
   ;I;")"
430 INPUT V1(I),V2(I)
440 NEXT I
450 DISP "DRAPED SURVEY? (Y/N)"
460 INPUT Y0$
470 IF Y0$#"Y" THEN 680
480 DISP "DRAPED DIST." ! **DRAP
   E
490 INPUT V3
500 FOR M=1 TO M0
510 W=X(M)
520 REM ***INTERPOLATION***
530 IF W>V1(1) THEN 560
540 F0=V2(1)
550 GOTO 630
560 IF W<V1(V) THEN 590
570 F0=V2(V)
580 GOTO 630
590 FOR J=2 TO V
600 IF W<V1(J) THEN 620
610 NEXT J
620 F0=V2(J-1)+(V2(J)-V2(J-1))*(
   W-V1(J-1))/(V1(J)-V1(J-1))
630 Z(M)=F0+V3
640 NEXT M
650 GOTO 750
660 V=1 ! ***CONST ALT SURVEYS
670 V1(1)=X8
680 DISP "FIELDZ"
690 INPUT E
700 FOR M=1 TO M0
710 Z(M)=E
720 NEXT M
730 IF V=1 THEN V2(1)=E
740 V3=0
750 ! ***BODY PARAMS**
760 DISP "HOW MANY BODIES?"
770 INPUT N
780 IF N>0 THEN 1030
790 DISP "TOPO DENS?"
800 INPUT K(1)
810 L0=99999
820 FOR I=1 TO V
830 X0(I,1)=V1(I)
840 Z0(I,1)=V2(I)
850 IF V2(I)<L0 THEN L0=V2(I)
860 NEXT I
870 I=0
880 IF V2(V)=L0 THEN 940
890 X0(V+1,1)=2*X9-X8
900 Z0(V+1,1)=V2(V)
910 X0(V+2,1)=2*X9-X8
920 Z0(V+2,1)=L0
930 I=2
940 IF V2(1)=L0 THEN 1000
950 X0(V+I+1,1)=2*X8-X9
960 Z0(V+I+1,1)=L0
970 X0(V+I+2,1)=2*X8-X9
980 Z0(V+I+2,1)=V2(1)
990 I=I+2
1000 N=1
1010 N1(1)=V+I
1020 GOTO 1140
1030 FOR I=1 TO N
1040 DISP "FOR BODY NO.";I
1050 DISP "DENS?"
1060 INPUT K(I)
1070 DISP "N CORNERS"

```

```

1080 INPUT N1(I)
1090 FOR J=1 TO N1(I)
1100 DISP "BODX(";J;"),BODZ(";J;")
1110 INPUT X0(J,I),Z0(J,I)
1120 NEXT J
1130 NEXT I
1140 GOSUB 2320 ! ***DRAW PICTURE
1150 DISP "PRINT BODS? (Y/N)"
1160 INPUT Y0$
1170 IF Y0$#"Y" THEN 1270
1180 FOR I=1 TO N
1190 PRINT
1200 PRINT "BODY #";I;"
      DENS=";K(I)
1210 PRINT " J      BODX      BODZ
      "
1220 FOR J=1 TO N1(I)
1230 IMAGE DD,2(X,DDDD.DDD)
1240 PRINT USING 1230 ; J;X0(J,I);Z0(J,I)
1250 NEXT J
1260 NEXT I
1270 RAD ! ***CALCULATE FIELD**
1280 FOR M=1 TO M0
1290 F(M)=0
1300 NEXT M
1310 FOR I=1 TO N
1320 N0=N1(I)
1330 DISP
1340 DISP "***FOR BODY #";I
1350 DISP " FIELDX FIELDZ FIELD"
1360 P1=0 @ P2=0
1370 FOR M=1 TO M0
1380 X1=X0(N0,I)-X(M)
1390 Z1=-Z0(N0,I)+Z(M)
1400 R1=SQR(X1^2+Z1^2)
1410 IF R1=0 THEN 1430
1420 R1=LOG(R1)
1430 IF X1=0 AND Z1=0 THEN 1450
1440 P1=ATN2(X1,Z1)
1450 E=0
1460 FOR J=1 TO N0
1470 X2=X0(J,I)-X(M)
1480 Z2=-Z0(J,I)+Z(M)
1490 R2=SQR(X2^2+Z2^2)
1500 IF R2=0 THEN 1520
1510 R2=LOG(R2)
1520 IF X2=0 AND Z2=0 THEN 1540
1530 P2=ATN2(X2,Z2)
1540 S=X2-X1
1550 C=Z2-Z1
1560 R6=(X2-X1)^2+(Z2-Z1)^2
1570 L=R2-R1
1580 T2=P2-P1
1590 IF T2<PI THEN T2=T2+2*PI
1600 IF T2>PI THEN T2=T2-2*PI
1610 B=(C*L+S*T2)*(X1*Z2-Z1*X2)/R6
1620 E=E+B
1630 X1=X2
1640 Z1=Z2
1650 R1=R2
1660 P1=P2
1670 NEXT J
1680 T=2*G*K(I)*E
1690 F(M)=F(M)+T
1700 DISP USING 110 ; X(M),Z(M),T
1710 NEXT M
1720 NEXT I
1730 DISP "PRINT OUTPUT? (Y/N)"
1740 INPUT Y0$
1750 IF Y0$#"Y" THEN 1820
1760 PRINT
1770 PRINT "SUM FIELD FOR ALL BODIES"
1780 PRINT " FIELDX FIELDZ FIELD"
1790 FOR M=1 TO M0
1800 PRINT USING 110 ; X(M),Z(M),F(M)
1810 NEXT M
1820 GOSUB 2720 ! **PLOT FIELDS
1830 PAUSE
1840 ALPHA
1850 DISP "TAPE THIS CASE? (Y/N)"
1860 INPUT Y0$
1870 IF Y0$#"Y" THEN 1890
1880 GOSUB 3010 ! **TAPE MAKER
1890 DISP "TRY CHANGES? (Y/N)"
1900 INPUT Y0$
1910 IF Y0$#"N" THEN 1940
1920 DISP "'BYE.....'"
1930 END
1940 DISP "CHANGE BODY CORNERS? (Y/N)"
1950 INPUT Y0$
1960 IF Y0$#"Y" THEN 2180
1970 FOR I=1 TO N
1980 DISP "BODY #";I
1990 DISP " (TYPE 0 TO ESCAPE THIS BODY.)"
2000 DISP "CHANGE CORNER #"
2010 INPUT J
2020 IF J=0 THEN 2170
2030 IF J<0 THEN 2080
2040 IF J>N1(I) THEN N1(I)=J
2050 DISP "NEWX(";J;"),NEWZ(";J;")
2060 INPUT X0(J,I),Z0(J,I)
2070 GOTO 2000
2080 J=-J ! DELETE THIS CORNER
2090 IF J=N1(I) THEN 2150
2100 IF J>N1(I) THEN 2170
2110 FOR M=J TO N1(I)-1
2120 X0(M,I)=X0(M+1,I)
2130 Z0(M,I)=Z0(M+1,I)

```

```

2140 NEXT M
2150 N1(I)=N1(I)-1
2160 GOTO 2000
2170 NEXT I
2180 DISP "CHANGE DENS? (Y/N)"
2190 INPUT Y0$
2200 IF Y0$#"Y" THEN 2250
2210 FOR I=1 TO N
2220 DISP "DENS FOR BODY#";I
2230 INPUT K(I)
2240 NEXT I
2250 DISP "ADD BOOS? (Y,N)"
2260 INPUT Y0$
2270 IF Y0$#"Y" THEN 1140
2280 N=N+1
2290 I=N
2300 FOR I=N TO N
2310 GOTO 1040
2320 GRAPH ! ***DRAW PICTURE***
2330 PEN 1 @ GCLEAR
2340 H1=-99999
2350 FOR M=1 TO M0
2360 IF Z(M)>H1 THEN H1=Z(M)
2370 NEXT M
2380 X7=X9-X8
2390 Y8=H1-X7/4
2400 Y9=H1+X7/2
2410 SCALE X8,X9,Y8,Y9
2420 PEN 1
2430 MOVE X9,V2(V)
2440 DRAW X9,Y8
2450 DRAW X8,Y8
2460 DRAW X8,V2(1)
2470 FOR I=1 TO V
2480 DRAW V1(I),V2(I)
2490 NEXT I
2500 DRAW X9,V2(V)
2510 PENUP
2520 IF N<=0 THEN RETURN
2530 FOR M=2 TO M0 STEP 2
2540 PLOT X(M-1),Z(M-1)
2550 PLOT X(M),Z(M)
2560 PENUP
2570 NEXT M
2580 PENUP
2590 FOR I=1 TO N
2600 N0=N1(I)
2610 MOVE X0(N0,I),Z0(N0,I)
2620 FOR J=1 TO N0
2630 DRAW X0(J,I),Z0(J,I)
2640 NEXT J
2650 NEXT I
2660 PAUSE
2670 ALPHA
2680 DISP "OKAY? (Y/N)"
2690 INPUT Y0$
2700 IF Y0$#"N" THEN 1940
2710 RETURN
2720 ! ***PLOT FIELDS***
2730 H1=-99999

```

```

2740 L0=99999
2750 FOR M=1 TO M0
2760 IF F(M)<L0 THEN L0=F(M)
2770 IF F(M)>H1 THEN H1=F(M)
2780 NEXT M
2790 DISP "FMIN (<;L0;; FMAX >";H1
2800 INPUT Z8,Z9
2810 Z7=Z9-Z8
2820 Y8=Z9-1.5*Z7
2830 GRAPH @ PEN 1
2840 SCALE X8,X9,Y8,Z9
2850 XAXIS 0,0
2860 YAXIS X8,Z7/10,Z8,Z9
2870 Z5=(Z9-Y8)/(2*M0)
2880 FOR M=1 TO M0
2890 MOVE X(M),F(M)
2900 IMOVE D/4,Z5/2
2910 IDRAW -(D/2),-Z5
2920 IMOVE 0,Z5
2930 IDRAW D/2,-Z5
2940 NEXT M
2950 IF H(1)=0 THEN RETURN
2960 MOVE X(1),H(1)
2970 FOR M=1 TO M0
2980 DRAW X(M),H(M)
2990 NEXT M
3000 RETURN
3010 ! ***TAPE MAKER
3020 FOR I=1 TO N ! ADD Y1,Y2
3030 Y1(I)=-5000 ! TO FILL OUT
3040 Y2(I)=5000 ! TAPE FORMAT
3050 NEXT I
3060 DISP "INSERT DATA TAPE IN CARRIER"
3070 DISP "FILE NAME? (6 CHARS)"
3080 INPUT B$
3090 CREATE B$,23,240
3100 ASSIGN# 1 TO B$
3110 PRINT# 1 ; X8,X9,D,V3,M0,V,N,G,G,G
3120 PRINT# 1 ; N1(),K(),X(),Z(),V1(),V2(),X0(),Z0(),H(),Y1(),Y2(),F()
3130 ASSIGN# 1 TO *
3140 RETURN
3150 ! ***TAPE READER
3160 DISP "INSERT DATA TAPE IN CARRIER"
3170 DISP "FILE NAME? (6 CHARS)"
3180 INPUT B$
3190 ASSIGN# 1 TO B$
3200 READ# 1 ; X8,X9,D,V3,M0,V,N,G,G,G
3210 READ# 1 ; N1(),K(),X(),Z(),V1(),V2(),X0(),Z0(),H()
3220 ASSIGN# 1 TO *
3230 GOTO 1140

```

```

10 ! PROGRAM 2HDGRV**SEPT,1982*
   **DAVE CAMPBELL, USGS-DENVER
20 CLEAR
30 DISP "2HDGRV FORWARD CALC'N"
40 OPTION BASE 1
50 DEG
60 INTEGER I,J,M,M0,N,N0,N1(5),
   V
70 SHORT X0(20,5),Z0(20,5),X(10
   0),Z(100),F(100),V1(20),V2(2
   0),K(5),Y1(5),Y2(5),H(100)
80 SHORT G,X8,X9,D,V3,W,F0,X1,Z
   1,R1,P1,Q1,X2,Z2,R2,P2,Q2,T
90 REAL U1,U2,W1,E1,E2,E,B,P,S,
   C,T2,L
100 SHORT Y8,Y9,Z7,X7,H1,L0,Z8,Z
   9,Z5 ! VARIABLES USED IN SUB
   S
110 IMAGE 2(DDDD.DDD,X),DDDD.DD
120 G=0 @ H(1)=0
130 DISP "IS DATA ON TAPE? (Y/N)
   "
140 INPUT Y0$
150 IF Y0$="Y" THEN 3330
160 DISP "KFT OR KM?"
170 INPUT Y0$
180 IF Y0$="KFT" THEN G=2.032
190 IF Y0$="KM" THEN G=6.667
200 IF G=0 THEN GOTO 160
210 DISP "***SET-UP FIELD POINTS
   ***"
220 DISP "FIELDX: XMIN,XMAX,DX"
230 INPUT X8,X9,D
240 M0=1+INT((X9-X8)/D)
250 FOR M=1 TO M0
260 X(M)=X8+(M-1)*D
270 NEXT M
280 DISP "INPUT OBS FIELDS? (Y/N
   )"
290 INPUT Y0$
300 IF Y0$#"Y" THEN 350
310 FOR M=1 TO M0
320 DISP "OBS( ";X(M);")"
330 INPUT H(M)
340 NEXT M
350 DISP "INPUT TOPO? (Y/N)"
360 INPUT Y0$
370 IF Y0$#"Y" THEN 660
380 ! ***TOPO INPUT**
390 DISP "HOW MANY TOPO POINTS"
400 INPUT V
410 FOR I=1 TO V
420 DISP "TOPO-X( ";I;"),TOPO-Z(
   ";I;")"
430 INPUT V1(I),V2(I)
440 NEXT I
450 DISP "DRAPED SURVEY? (Y/N)"
460 INPUT Y0$
470 IF Y0$#"Y" THEN 680
480 DISP "DRAPED DIST." ! **DRAP
   E
490 INPUT V3
500 FOR M=1 TO M0
510 W=X(M)
520 REM ***INTERPOLATION***
530 IF W>V1(1) THEN 560
540 F0=V2(1)
550 GOTO 630
560 IF W<V1(V) THEN 590
570 F0=V2(V)
580 GOTO 630
590 FOR J=2 TO V
600 IF W<V1(J) THEN 620
610 NEXT J
620 F0=V2(J-1)+(V2(J)-V2(J-1))*(
   W-V1(J-1))/(V1(J)-V1(J-1))
630 Z(M)=F0+V3
640 NEXT M
650 GOTO 750
660 V=1 ! **CONST ALT SURVEYS
670 V1(1)=X8
680 DISP "FIELD Z"
690 INPUT E
700 FOR M=1 TO M0
710 Z(M)=E
720 NEXT M
730 IF V=1 THEN V2(1)=E
740 V3=0
750 ! **BODY PARAMS**
760 DISP "HOW MANY BODIES?"
770 INPUT N
780 IF N>0 THEN 1050
790 DISP "TOPO DENS?"
800 INPUT K(1)
810 DISP "TOPO HALF-LENGTH?"
820 INPUT Y2(1)
830 L0=99999
840 FOR I=1 TO V
850 X0(I,1)=V1(I)
860 Z0(I,1)=V2(I)
870 IF V2(I)<L0 THEN L0=V2(I)
880 NEXT I
890 I=0
900 IF V2(V)=L0 THEN 960
910 X0(V+1,1)=2*X9-X8
920 Z0(V+1,1)=V2(V)
930 X0(V+2,1)=2*X9-X8
940 Z0(V+2,1)=L0
950 I=2
960 IF V2(1)=L0 THEN 1020
970 X0(V+I+1,1)=2*X8-X9
980 Z0(V+I+1,1)=L0
990 X0(V+I+2,1)=2*X8-X9
1000 Z0(V+I+2,1)=V2(1)
1010 I=I+2
1020 N=1
1030 N1(1)=V+I
1040 GOTO 1180
1050 FOR I=1 TO N
1060 DISP "FOR BODY NO. ";I
1070 DISP "DENS?"

```

```

1080 INPUT K(I)
1090 DISP "N CORNERS"
1100 INPUT N1(I)
1110 FOR J=1 TO N1(I)
1120 DISP "BODX(";J;"),BODZ(";J;
      ")"
1130 INPUT X0(J,I),Z0(J,I)
1140 NEXT J
1150 DISP "HALF-LENGTH Y(";I;)"
1160 INPUT Y2(I)
1170 NEXT I
1180 GOSUB 2510 ! **DRAW PICTURE
1190 DISP "PRINT BODS? (Y/N)"
1200 INPUT Y0$
1210 IF Y0$#"Y" THEN 1320
1220 FOR I=1 TO N
1230 PRINT
1240 PRINT "BODY #";I;
      DENS="";K(I)
1250 PRINT "                HALF-WID
      TH=";Y2(I)
1260 PRINT " J          BODX          BODZ
      "
1270 FOR J=1 TO N1(I)
1280 IMAGE DD,2(X,DDDD.DDD)
1290 PRINT USING 1280 ; J;X0(J,I)
      ;Z0(J,I)
1300 NEXT J
1310 NEXT I
1320 RAD ! ***CALCULATE FIELD**
1330 FOR M=1 TO M0
1340 F(M)=0
1350 NEXT M
1360 FOR I=1 TO N
1370 N0=N1(I)
1380 Q=Y2(I)
1390 DISP
1400 DISP "***FOR BODY #";I
1410 DISP " FIELDX FIELDZ
      FIELD"
1420 FOR M=1 TO M0
1430 X1=X0(N0,I)-X(M)
1440 Z1=-Z0(N0,I)+Z(M)
1450 R1=SQR(X1^2+Z1^2)
1460 P1=SQR(R1^2+Q^2)
1470 E1=R1/(P1+Q)
1480 IF E1=0 THEN 1500
1490 E1=LOG(E1)
1500 E=0
1510 FOR J=1 TO N0
1520 X2=X0(J,I)-X(M)
1530 Z2=-Z0(J,I)+Z(M)
1540 R2=SQR(X2^2+Z2^2)
1550 P2=SQR(R2^2+Q^2)
1560 E2=R2/(P2+Q)
1570 IF E2=0 THEN 1590
1580 E2=LOG(E2)
1590 L=E2-E1
1600 R6=SQR((X2-X1)^2+(Z2-Z1)^2)
1610 P=ATN2(Z2-Z1,X2-X1)
1620 C=COS(P)
1630 S=SIN(P)
1640 U1=C*X1+S*Z1
1650 U2=C*X2+S*Z2
1660 W1=C*Z1-S*X1
1670 IF U2=0 AND W1=0 THEN 1720
1680 IF U1=0 AND W1=0 THEN 1720
1690 T2=ATN2(U2*Q,W1*P2)-ATN2(U1
      *Q,W1*P1)
1700 IF T2<-PI THEN T2=T2+2*PI
1710 IF T2>PI THEN T2=T2-2*PI
1720 B=(S*L+C*T2)*(X1*Z2-Z1*X2)/
      R6
1730 B=B+Q*C*LOG((U2+P2)/(U1+P1)
      )
1740 E=E+B
1750 X1=X2
1760 Z1=Z2
1770 R1=R2
1780 P1=P2
1790 E1=E2
1800 NEXT J
1810 T=2*G*K(I)*E
1820 F(M)=F(M)+T
1830 DISP USING 110 ; X(M),Z(M),
      T
1840 NEXT M
1850 NEXT I
1860 DISP "PRINT OUTPUT? (Y/N)"
1870 INPUT Y0$
1880 IF Y0$#"Y" THEN 1950
1890 PRINT
1900 PRINT "SUM FIELD FOR ALL BO
      DIES"
1910 PRINT " FIELDX FIELDZ
      FIELD"
1920 FOR M=1 TO M0
1930 PRINT USING 110 ; X(M),Z(M)
      ,F(M)
1940 NEXT M
1950 GOSUB 2910 ! **PLOT FIELDS
1960 PAUSE
1970 ALPHA
1980 DISP "TAPE THIS CASE? (Y/N)"
      "
1990 INPUT Y0$
2000 IF Y0$#"Y" THEN 2020
2010 GOSUB 3200 ! **TAPE MAKER
2020 DISP "TRY CHANGES? (Y/N)"
2030 INPUT Y0$
2040 IF Y0$#"N" THEN 2070
2050 DISP "'BYE.....'"
2060 END
2070 DISP "CHANGE BODY CORNERS?
      (Y/N)"
2080 INPUT Y0$
2090 FOR I=1 TO N
2100 DISP "BODY #";I
2110 DISP " (TYPE 0 TO ESCAPE T
      HIS BODY.)"

```

```

2120 DISP "CHANGE CORNER #"
2130 INPUT J
2140 IF J=0 THEN 2290
2150 IF J<0 THEN 2200
2160 IF J>N1(I) THEN N1(I)=J
2170 DISP "NEWX(";J;"),NEWZ(";J;
      ")"
2180 INPUT X0(J,I),Z0(J,I)
2190 GOTO 2120
2200 J=-J ! DELETE CORNER J
2210 IF J=N1(I) THEN 2270
2220 IF J>N1(I) THEN 2290
2230 FOR M=J TO N1(I)-1
2240 X0(M,I)=X0(M+1,I)
2250 Z0(M,I)=Z0(M+1,I)
2260 NEXT M
2270 N1(I)=N1(I)-1
2280 GOTO 2120
2290 NEXT I
2300 DISP "CHANGE DENS? (Y/N)"
2310 INPUT Y0$
2320 IF Y0$#"Y" THEN 2370
2330 FOR I=1 TO N
2340 DISP "DENS FOR BODY#";I
2350 INPUT K(I)
2360 NEXT I
2370 DISP "CHANGE STRIKE-LENGTHS
      ? (Y/N)"
2380 INPUT Y0$
2390 IF Y0$#"Y" THEN 2440
2400 FOR I=1 TO N
2410 DISP "Y FOR BODY#";I
2420 INPUT Y2(I)
2430 NEXT I
2440 DISP "ADD BODS? (Y,N)"
2450 INPUT Y0$
2460 IF Y0$#"Y" THEN 1180
2470 N=N+1
2480 I=N
2490 FOR I=N TO N
2500 GOTO 1060
2510 GRAPH ! ***DRAW PICTURE***
2520 PEN 1 @ GCLEAR
2530 H1=-99999
2540 FOR M=1 TO M0
2550 IF Z(M)>H1 THEN H1=Z(M)
2560 NEXT M
2570 X7=X9-X8
2580 Y8=H1-X7/4
2590 Y9=H1+X7/2
2600 SCALE X8,X9,Y8,Y9
2610 PEN 1
2620 MOVE X9,V2(V)
2630 DRAW X9,Y8
2640 DRAW X8,Y8
2650 DRAW X8,V2(1)
2660 FOR I=1 TO V
2670 DRAW V1(I),V2(I)
2680 NEXT I
2690 DRAW X9,V2(V)
2700 PENUP
2710 IF N<=0 THEN RETURN
2720 FOR M=2 TO M0 STEP 2
2730 PLOT X(M-1),Z(M-1)
2740 PLOT X(M),Z(M)
2750 PENUP
2760 NEXT M
2770 PENUP
2780 FOR I=1 TO N
2790 N0=N1(I)
2800 MOVE X0(N0,I),Z0(N0,I)
2810 FOR J=1 TO N0
2820 DRAW X0(J,I),Z0(J,I)
2830 NEXT J
2840 NEXT I
2850 PAUSE
2860 ALPHA
2870 DISP "OKAY? (Y/N)"
2880 INPUT Y0$
2890 IF Y0$="N" THEN 2070
2900 RETURN
2910 ! ***PLOT FIELDS***
2920 H1=-99999
2930 L0=99999
2940 FOR M=1 TO M0
2950 IF F(M)<L0 THEN L0=F(M)
2960 IF F(M)>H1 THEN H1=F(M)
2970 NEXT M
2980 DISP "FMIN (<;L0;; FMAX >"
      ";H1
2990 INPUT Z8,Z9
3000 Z7=Z9-Z8
3010 Y8=Z9-1.5*Z7
3020 GRAPH @ PEN 1
3030 SCALE X8,X9,Y8,Z9
3040 XAXIS 0,0
3050 YAXIS X8,Z7/10,Z8,Z9
3060 Z5=(Z9-Z8)/(2*M0)
3070 FOR M=1 TO M0
3080 MOVE X(M),F(M)
3090 IMOVE D/4,Z5/2
3100 IDRAW -(D/2),-Z5
3110 IMOVE 0,Z5
3120 IDRAW D/2,-Z5
3130 NEXT M
3140 IF H(1)=0 THEN RETURN
3150 MOVE X(1),H(1)
3160 FOR M=1 TO M0
3170 DRAW X(M),H(M)
3180 NEXT M
3190 RETURN
3200 ! ***TAPE MAKER
3210 FOR I=1 TO N ! ADD Y1'S TO
3220 Y1(I)=-Y2(I) ! FILL OUT
3230 NEXT I ! TAPE FORMAT
3240 DISP "INSERT DATA TAPE IN C
      ARRIER"
3250 DISP "FILE NAME? (6 CHARS)"
3260 INPUT B$
3270 CREATE B$,23,240

```

```
3280 ASSIGN# 1 TO B$
3290 PRINT# 1 ; X8,X9,D,V3,M0,V,
    N,G,G,G
3300 PRINT# 1 ; N1(),K(),X(),Z(),
    V1(),V2(),X0(),Z0(),H(),
    Y1(),Y2(),F()
3310 ASSIGN# 1 TO *
3320 RETURN
3330 ! ***TAPE READER
3340 DISP "INSERT DATA TAPE IN C
    ARRIER"
3350 DISP "FILE NAME? (6 CHARS)"
3360 INPUT B$
3370 ASSIGN# 1 TO B$
3380 READ# 1 ; X8,X9,D,V3,M0,V,N
    ,G,G,G
3390 READ# 1 ; N1(),K(),X(),Z(),
    V1(),V2(),X0(),Z0(),H(),Y
    1(),Y2()
3400 ASSIGN# 1 TO *
3410 GOTO 1180
```

```

10 ! PROGRAM NEWGRV**SEPT,1982*
   **DAVE CAMPBELL, USGS-DENVER
20 CLEAR
30 DISP "2HGRV FORWARD CALC'N"
40 OPTION BASE 1
50 DEG
60 INTEGER I,J,M,M0,N,N0,N1(5),
   V
70 SHORT X0(20,5),Z0(20,5),X(10
   0),Z(100),F(100),V1(20),V2(2
   0),K(5),Y1(5),Y2(5),H(100)
80 SHORT G,X8,X9,D,V3,W,F0,X1,Z
   1,R1,P1,Q1,X2,Z2,R2,P2,Q2,T
90 REAL U1,U2,W1,E1,E2,E3,P,S,C
   ,T2,T3,L,B,E
100 SHORT Y8,Y9,Z7,X7,H1,L0,Z8,Z
   9,Z5 ! VARIABLES USED IN SUB
   S
110 IMAGE 2(DDDD.DDD,X),DDDDD.DD
120 G=0 @ H(1)=0
130 DISP "IS DATA ON TAPE? (Y/N)
   "
140 INPUT Y0$
150 IF Y0$="Y" THEN 3450
160 DISP "KFT OR KM?"
170 INPUT Y0$
180 IF Y0$="KFT" THEN G=2.032
190 IF Y0$="KM" THEN G=6.667
200 IF G=0 THEN GOTO 160
210 DISP "***SET-UP FIELD POINTS
   ***"
220 DISP "FIELDX: XMIN,XMAX,DX"
230 INPUT X8,X9,D
240 M0=1+INT((X9-X8)/D)
250 FOR M=1 TO M0
260 X(M)=X8+(M-1)*D
270 NEXT M
280 DISP "INPUT OBS FIELDS? (Y/N
   )"
290 INPUT Y0$
300 IF Y0$#"Y" THEN 350
310 FOR M=1 TO M0
320 DISP "OBS(";X(M);")"
330 INPUT H(M)
340 NEXT M
350 DISP "INPUT TOPO? (Y/N)"
360 INPUT Y0$
370 IF Y0$#"Y" THEN 660
380 ! **TOPO INPUT**
390 DISP "HOW MANY TOPO POINTS"
400 INPUT V
410 FOR I=1 TO V
420 DISP "TOPO-X(";I;"),TOPO-Z("
   ;I;")"
430 INPUT V1(I),V2(I)
440 NEXT I
450 DISP "DRAPED SURVEY? (Y/N)"
460 INPUT Y0$
470 IF Y0$#"Y" THEN 680
480 DISP "DRAPED DIST." ! **DRAP
   E
490 INPUT V3
500 FOR M=1 TO M0
510 W=X(M)
520 REM ***INTERPOLATION***
530 IF W>V1(1) THEN 560
540 F0=V2(1)
550 GOTO 630
560 IF W<V1(V) THEN 590
570 F0=V2(V)
580 GOTO 630
590 FOR J=2 TO V
600 IF W<V1(J) THEN 620
610 NEXT J
620 F0=V2(J-1)+(V2(J)-V2(J-1))*(
   W-V1(J-1))/(V1(J)-V1(J-1))
630 Z(M)=F0+V3
640 NEXT M
650 GOTO 750
660 V=1 ! **CONST ALT SURVEYS
670 V1(1)=X8
680 DISP "FIELD Z"
690 INPUT E
700 FOR M=1 TO M0
710 Z(M)=E
720 NEXT M
730 IF V=1 THEN V2(1)=E
740 V3=0
750 ! **BODY PARAMS**
760 DISP "HOW MANY BODIES?"
770 INPUT N
780 IF N>0 THEN 1050
790 DISP "TOPO DENS?"
800 INPUT K(1)
810 DISP "TOPO END COORDS Y1,Y2?
   "
820 INPUT Y1(1),Y2(1)
830 L0=99999
840 FOR I=1 TO V
850 X0(I,1)=V1(I)
860 Z0(I,1)=V2(I)
870 IF V2(I)<L0 THEN L0=V2(I)
880 NEXT I
890 I=0
900 IF V2(V)=L0 THEN 960
910 X0(V+1,1)=2*X9-X8
920 Z0(V+1,1)=Z0(V,1)
930 X0(V+2,1)=2*X9-X8
940 Z0(V+2,1)=L0
950 I=2
960 IF V2(1)=L0 THEN 1020
970 X0(V+3,1)=2*X8-X9
980 Z0(V+3,1)=L0
990 X0(V+4,1)=2*X8-X9
1000 Z0(V+4,1)=Z0(1,1)
1010 I=I+2
1020 N=1
1030 N1(1)=V+I
1040 GOTO 1180
1050 FOR I=1 TO N
1060 DISP "FOR BODY NO.";I

```



```

1070 DISP "DENS?"
1080 INPUT K(I)
1090 DISP "N CORNERS"
1100 INPUT N1(I)
1110 FOR J=1 TO N1(I)
1120 DISP "BODX(";J;"),BODZ(";J;")
1130 INPUT X0(J,I),Z0(J,I)
1140 NEXT J
1150 DISP "HALF-LENGTHS Y1(";I;")
      Y2(";I;")"
1160 INPUT Y1(I),Y2(I)
1170 NEXT I
1180 GOSUB 2660 ! **DRAW PICTURE
1190 DISP "PRINT BODS? (Y/N)"
1200 INPUT Y0$
1210 IF Y0$#"Y" THEN 1320
1220 FOR I=1 TO N
1230 PRINT
1240 PRINT "BODY #";I;"
      DENS=";K(I)
1250 PRINT " Y1=";Y1(I);" Y2="
      Y2(I)
1260 PRINT " J      BODX      BODZ
      "
1270 FOR J=1 TO N1(I)
1280 IMAGE DD,2(X,DDDD.DDD)
1290 PRINT USING 1280 ; J;X0(J,I)
      Z0(J,I)
1300 NEXT J
1310 NEXT I
1320 RAD ! **CALCULATE FIELD**
1330 FOR M=1 TO M0
1340 F(M)=0
1350 NEXT M
1360 FOR I=1 TO N
1370 N0=N1(I)
1380 DISP
1390 DISP "***FOR BODY #";I
1400 DISP " FIELDX FIELDZ
      FIELD"
1410 FOR M=1 TO M0
1420 X1=X0(N0,I)-X(M)
1430 Z1=-Z0(N0,I)+Z(M)
1440 R1=SQR(X1^2+Z1^2)
1450 P1=SQR(R1^2+Y1(I)^2)
1460 Q1=SQR(R1^2+Y2(I)^2)
1470 E1=R1/(Q1+Y2(I))
1480 IF E1=0 THEN 1500
1490 E1=LOG(E1)
1500 E3=R1/(P1-Y1(I))
1510 IF E3=0 THEN 1530
1520 E3=LOG(E3)
1530 E1=E1+E3
1540 E=0
1550 FOR J=1 TO N0
1560 X2=X0(J,I)-X(M)
1570 Z2=-Z0(J,I)+Z(M)
1580 R2=SQR(X2^2+Z2^2)
1590 P2=SQR(R2^2+Y1(I)^2)
1600 Q2=SQR(R2^2+Y2(I)^2)
1610 E2=R2/(Q2+Y2(I))
1620 IF E2=0 THEN 1640
1630 E2=LOG(E2)
1640 E3=R2/(P2-Y1(I))
1650 IF E3=0 THEN 1670
1660 E3=LOG(E3)
1670 E2=E2+E3
1680 L=E2-E1
1690 R6=SQR((X2-X1)^2+(Z2-Z1)^2)
1700 P=ATN2(Z2-Z1,X2-X1)
1710 C=COS(P)
1720 S=SIN(P)
1730 U1=C*X1+S*Z1
1740 U2=C*X2+S*Z2
1750 W1=C*Z1-S*X1
1760 IF U2=0 AND W1=0 THEN 1850
1770 IF U1=0 AND W1=0 THEN 1850
1780 T2=ATN2(U2*Y2(I),W1*Q2)-ATN
      2(U1*Y2(I),W1*Q1)
1790 IF T2<-PI THEN T2=T2+2*PI
1800 IF T2>PI THEN T2=T2-2*PI
1810 T3=ATN2(U2*Y1(I),W1*P2)-ATN
      2(U1*Y1(I),W1*P1)
1820 IF T3<-PI THEN T3=T3+2*PI
1830 IF T3>PI THEN T3=T3-2*PI
1840 T2=T2-T3
1850 B=(S*L+C*T2)*(X1*Z2-Z1*X2)/
      R6
1860 B=B+Y2(I)*C*LOG((U2+Q2)/(U1
      +Q1))
1870 B=B-Y1(I)*C*LOG((U2+P2)/(U1
      +P1))
1880 E=E+B
1890 X1=X2
1900 Z1=Z2
1910 R1=R2
1920 P1=P2
1930 Q1=Q2
1940 E1=E2
1950 NEXT J
1960 T=G*K(I)*E
1970 F(M)=F(M)+T
1980 DISP USING 110 ; X(M),Z(M),
      T
1990 NEXT M
2000 NEXT I
2010 DISP "PRINT OUTPUT? (Y/N)"
2020 INPUT Y0$
2030 IF Y0$#"Y" THEN 2100
2040 PRINT
2050 PRINT "SUM FIELD FOR ALL BO
      DIES"
2060 PRINT " FIELDX FIELDZ
      FIELD"
2070 FOR M=1 TO M0
2080 PRINT USING 110 ; X(M),Z(M)
      ,F(M)
2090 NEXT M
2100 GOSUB 3060 ! **PLOT FIELDS

```

```

2110 PAUSE
2120 ALPHA
2130 DISP "TAPE THIS CASE? (Y/N)"
2140 INPUT Y0$
2150 IF Y0$#"Y" THEN 2170
2160 GOSUB 3350 ! **TAPE MAKER
2170 DISP "TRY CHANGES? (Y/N)"
2180 INPUT Y0$
2190 IF Y0$#"N" THEN 2220
2200 DISP "'BYE.....'"
2210 END
2220 DISP "CHANGE BODY CORNERS?
(Y/N)"
2230 INPUT Y0$
2240 FOR I=1 TO N
2250 DISP "BODY #";I
2260 DISP " (TYPE 0 TO ESCAPE T
HIS BODY.)"
2270 DISP "CHANGE CORNER #"
2280 INPUT J
2290 IF J=0 THEN 2440
2300 IF J<0 THEN 2350
2310 IF J>N1(I) THEN N1(I)=J
2320 DISP "NEWX(";J;"),NEWZ(";J;
")"
2330 INPUT X0(J,I),Z0(J,I)
2340 GOTO 2270
2350 J=-J ! DELETE CORNER
2360 IF J=N1(I) THEN 2420
2370 IF J>N1(I) THEN 2440
2380 FOR M=J TO N1(I)-1
2390 X0(M,I)=X0(M+1,I)
2400 Z0(M,I)=Z0(M+1,I)
2410 NEXT M
2420 N1(I)=N1(I)-1
2430 GOTO 2270
2440 NEXT I
2450 DISP "CHANGE DENS? (Y/N)"
2460 INPUT Y0$
2470 IF Y0$#"Y" THEN 2520
2480 FOR I=1 TO N
2490 DISP "DENS OF BODY#";I
2500 INPUT K(I)
2510 NEXT I
2520 DISP "CHANGE END COORDS Y1,
Y2? (Y/N)"
2530 INPUT Y0$
2540 IF Y0$#"Y" THEN 2590
2550 FOR I=1 TO N
2560 DISP "Y1,Y2 FOR BODY#";I
2570 INPUT Y1(I),Y2(I)
2580 NEXT I
2590 DISP "ADD BODS? (Y,N)"
2600 INPUT Y0$
2610 IF Y0$#"Y" THEN 1180
2620 N=N+1
2630 I=N
2640 FOR I=N TO N
2650 GOTO 1060

```

```

2660 GRAPH ! ***DRAW PICTURE***
2670 PEN 1 @ GCLEAR
2680 H1=-99999
2690 FOR M=1 TO M0
2700 IF Z(M)>H1 THEN H1=Z(M)
2710 NEXT M
2720 X7=X9-X8
2730 Y8=H1-X7/4
2740 Y9=H1+X7/2
2750 SCALE X8,X9,Y8,Y9
2760 PEN 1
2770 MOVE X9,V2(V)
2780 DRAW X9,Y8
2790 DRAW X8,Y8
2800 DRAW X8,V2(I)
2810 FOR I=1 TO V
2820 DRAW V1(I),V2(I)
2830 NEXT I
2840 DRAW X9,V2(V)
2850 PENUP
2860 IF N<=0 THEN RETURN
2870 FOR M=2 TO M0 STEP 2
2880 PLOT X(M-1),Z(M-1)
2890 PLOT X(M),Z(M)
2900 PENUP
2910 NEXT M
2920 PENUP
2930 FOR I=1 TO N
2940 M0=N1(I)
2950 MOVE X0(N0,I),Z0(N0,I)
2960 FOR J=1 TO M0
2970 DRAW X0(J,I),Z0(J,I)
2980 NEXT J
2990 NEXT I
3000 PAUSE
3010 ALPHA
3020 DISP "OKAY? (Y/N)"
3030 INPUT Y0$
3040 IF Y0$#"N" THEN 2220
3050 RETURN
3060 ! ***PLOT FIELDS***
3070 H1=-99999
3080 L0=99999
3090 FOR M=1 TO M0
3100 IF F(M)<L0 THEN L0=F(M)
3110 IF F(M)>H1 THEN H1=F(M)
3120 NEXT M
3130 DISP "FMIN (<;L0;; FMAX >"
;H1
3140 INPUT Z8,Z9
3150 Z7=Z9-Z8
3160 Y8=Z9-1.5*Z7
3170 GRAPH @ PEN 1
3180 SCALE X8,X9,Y8,Z9
3190 XAXIS 0,0
3200 YAXIS X8,Z7/10,Z8,Z9
3210 Z5=(Z9-Z8)/(2*M0)
3220 FOR M=1 TO M0
3230 MOVE X(M),F(M)
3240 IMOVE 0/4,Z5/2

```

```

3250 IDRAW -(D/2),-Z5
3260 IMOVE 0,Z5
3270 IDRAW D/2,-Z5
3280 NEXT M
3290 IF H(1)=0 THEN RETURN
3300 MOVE X(1),H(1)
3310 FOR M=1 TO M0
3320 DRAW X(M),H(M)
3330 NEXT M
3340 RETURN
3350 ! ***TAPE MAKER
3360 DISP "INSERT DATA TAPE IN C
      ARRIER"
3370 DISP "FILE NAME? (6 CHARS)"
3380 INPUT B$
3390 CREATE B$,23,240
3400 ASSIGN# 1 TO B$
3410 PRINT# 1 ; X8,X9,D,V3,M0,V,
      N,G,G,G
3420 PRINT# 1 ; N1(),K(),X(),Z(),
      V1(),V2(),X0(),Z0(),H(),
      Y1(),Y2(),F()
3430 ASSIGN# 1 TO *
3440 RETURN
3450 ! ***TAPE READER
3460 DISP "INSERT DATA TAPE IN C
      ARRIER"
3470 DISP "FILE NAME? (6 CHARS)"
3480 INPUT B$
3490 ASSIGN# 1 TO B$
3500 READ# 1 ; X8,X9,D,V3,M0,V,N
      ,G,G,G
3510 READ# 1 ; N1(),K(),X(),Z(),
      V1(),V2(),X0(),Z0(),H(),Y
      1(),Y2()
3520 ASSIGN# 1 TO *
3530 GOTO 1180

```

```

10 ! PROGRAM 2DMAG***SEPT,1982*
   **DAVE CAMPBELL, USGS-DENVER
20 CLEAR
30 DISP "2DMAG FORWARD CALC'N"
40 OPTION BASE 1
50 DEG
60 INTEGER I,J,M,M0,N,N0,N1(5),
   V
70 SHORT X0(20,5),Z0(20,5),X(10
   0),Z(100),F(100),V1(20),V2(2
   0),K(5),Y1(5),Y2(5),H(100)
80 SHORT T0,I0,S0,H2,A,X8,X9,D,
   V3,W,F0,X1,Z1,R1,X2,Z2,R2
90 REAL P1,P2,B,C,D3,P3,E,T
100 SHORT Y8,Y9,Z7,X7,H1,L0,Z8,Z
   9,Z5 ! VARIABLES USED IN SUB
   S
110 IMAGE 2(DDDD.DDD,X),DDDD.DD
120 DISP "IS DATA ON TAPE? (Y/N)
   "
130 H(1)=0
140 INPUT Y0$
150 IF Y0$="Y" THEN 3080
160 DISP "TOTAL FIELD?"
170 INPUT T0
180 DISP "FIELD INCL (DEG)"
190 INPUT I0
200 DISP "STRIKE ANGLE (DEG)"
210 INPUT S0
220 DISP "***SET-UP FIELD POINTS
   ***"
230 DISP "FIELDX: XMIN,XMAX,DX"
240 INPUT X8,X9,D
250 M0=1+INT((X9-X8)/D)
260 FOR M=1 TO M0
270 X(M)=X8+(M-1)*D
280 NEXT M
290 DISP "INPUT OBS FIELDS? (Y/N
   )"
300 INPUT Y0$
310 IF Y0$#"Y" THEN 360
320 FOR M=1 TO M0
330 DISP "OBS(";X(M);")"
340 INPUT H(M)
350 NEXT M
360 DISP "INPUT TOPO? (Y/N)"
370 INPUT Y0$
380 IF Y0$#"Y" THEN 660
390 ! ***TOPO INPUT**
400 DISP "HOW MANY TOPO POINTS"
410 INPUT V
420 FOR I=1 TO V
430 DISP "TOPO-X(";I;"),TOPO-Z("
   ;I;")"
440 INPUT V1(I),V2(I)
450 NEXT I
460 DISP "DRAPED SURVEY? (Y/N)"
470 INPUT Y0$
480 IF Y0$#"Y" THEN 680
490 DISP "DRAPED DIST." ! **DRAP
   E
500 INPUT V3
510 FOR M=1 TO M0
520 W=X(M)
530 IF W>V1(1) THEN 560 ! **INTE
   RPOLATION**
540 F0=V2(1)
550 GOTO 630
560 IF W<V1(V) THEN 590
570 F0=V2(V)
580 GOTO 630
590 FOR J=2 TO V
600 IF W<V1(J) THEN 620
610 NEXT J
620 F0=V2(J-1)+(V2(J)-V2(J-1))*(
   W-V1(J-1))/(V1(J)-V1(J-1))
630 Z(M)=F0+V3
640 NEXT M
650 GOTO 750
660 V=1 ! ***CONST ALT SURVEYS
670 V1(1)=X8
680 DISP "FIELDZ"
690 INPUT E
700 FOR M=1 TO M0
710 Z(M)=E
720 NEXT M
730 IF V=1 THEN V2(1)=E
740 V3=0
750 ! ***BODY PARAMS**
760 DISP "HOW MANY BODIES?"
770 INPUT N
780 IF N>0 THEN 1030
790 DISP "TOPO SUSC."
800 INPUT K(1)
810 L0=99999
820 FOR I=1 TO V
830 X0(I,1)=V1(I)
840 Z0(I,1)=V2(I)
850 IF V2(I)<L0 THEN L0=V2(I)
860 NEXT I
870 I=0
880 IF V2(V)=L0 THEN 940
890 X0(V+1,1)=2*X9-X8
900 Z0(V+1,1)=V2(V)
910 X0(V+2,1)=2*X9-X8
920 Z0(V+2,1)=L0
930 I=2
940 IF V2(1)=L0 THEN 1000
950 X0(V+I+1,1)=2*X8-X9
960 Z0(V+I+1,1)=L0
970 X0(V+I+2,1)=2*X8-X9
980 Z0(V+I+2,1)=V2(1)
990 I=I+2
1000 N=N-1
1010 N1(1)=V+I
1020 GOTO 1140
1030 FOR I=1 TO N
1040 DISP "FOR BODY NO.";I
1050 DISP "SUSC."
1060 INPUT K(I)
1070 DISP "N CORNERS"

```

```

1080 INPUT N1(I)
1090 FOR J=1 TO N1(I)
1100 DISP "BODX(";J;"),BODZ(";J;
      ")"
1110 INPUT X0(J,I),Z0(J,I)
1120 NEXT J
1130 NEXT I
1140 GOSUB 2240 ! **DRAW PICTURE
1150 DISP "PRINT BODS? (Y/N)"
1160 INPUT Y0$
1170 IF Y0$#"Y" THEN 1290
1180 PRINT
1190 PRINT "FIELD=";T0;" INCL="
      ;I0;" STRIKE=";S0
1200 FOR I=1 TO N
1210 PRINT
1220 PRINT "BODY #";I;"
      SUSC=";K(I)
1230 PRINT " J      BODX      BODZ
      "
1240 FOR J=1 TO N1(I)
1250 IMAGE DD,2(X,DDDD.DDD)
1260 PRINT USING 1250 ; J,X0(J,I)
      ;Z0(J,I)
1270 NEXT J
1280 NEXT I
1290 RAD ! ***CALCULATE FIELD**
1300 FOR M=1 TO M0
1310 F(M)=0
1320 NEXT M
1330 H2=1-(COS(I0*PI/180)*COS(S0
      *PI/180))^2
1340 A=2*ATN(TAN(I0*PI/180)/SIN(
      S0*PI/180))
1350 FOR I=1 TO N
1360 N0=N1(I)
1370 DISP
1380 DISP "***FOR BODY #";I
1390 DISP " FIELDX FIELDZ
      FIELD"
1400 FOR M=1 TO M0
1410 X1=X0(N0,I)-X(M)
1420 Z1=-Z0(N0,I)+Z(M)
1430 R1=SQR(X1^2+Z1^2)
1440 P1=ATN2(X1,Z1)
1450 E=0
1460 FOR J=1 TO N0
1470 X2=X0(J,I)-X(M)
1480 Z2=-Z0(J,I)+Z(M)
1490 R2=SQR(X2^2+Z2^2)
1500 D3=LOG(R2/R1)
1510 P2=ATN2(X2,Z2)
1520 P3=P2-P1
1530 B=ATN2(Z2-Z1,X2-X1)
1540 C=A-B
1550 E=E-SIN(B)*(SIN(C)*P3+COS(C)
      )*D3)
1560 X1=X2
1570 Z1=Z2
1580 R1=R2
1590 P1=P2
1600 NEXT J
1610 T=2*T0*K(I)*H2*E
1620 F(M)=F(M)+T
1630 DISP USING 110 ; X(M),Z(M),
      T
1640 NEXT M
1650 NEXT I
1660 DISP "PRINT OUTPUT? (Y/N)"
1670 INPUT Y0$
1680 IF Y0$#"Y" THEN 1750
1690 PRINT
1700 PRINT "SUM FIELD FOR ALL BO
      DIES"
1710 PRINT " FIELDX FIELDZ
      FIELD"
1720 FOR M=1 TO M0
1730 PRINT USING 110 ; X(M),Z(M)
      ,F(M)
1740 NEXT M
1750 GOSUB 2650 ! **PLOT FIELDS
1760 PAUSE
1770 ALPHA
1780 DISP "TAPE THIS CASE? (Y/N)"
      "
1790 INPUT Y0$
1800 IF Y0$#"Y" THEN 1820
1810 GOSUB 2940 ! **TAPE MAKER
1820 DISP "TRY CHANGES? (Y/N)"
1830 INPUT Y0$
1840 IF Y0$#"N" THEN 1870
1850 DISP "'BYE.....'"
1860 END
1870 DISP "CHANGE BODY CORNERS?
      (Y/N)"
1880 INPUT Y0$
1890 IF Y0$#"Y" THEN 2110
1900 FOR I=1 TO N
1910 DISP "BODY #";I
1920 DISP " (TYPE 0 TO ESCAPE T
      HIS BODY.)"
1930 DISP "CHANGE CORNER #"
1940 INPUT J
1950 IF J=0 THEN 2100
1960 IF J<0 THEN 2010
1970 IF J>N1(I) THEN N1(I)=J
1980 DISP "NEWX(";J;"),NEWZ(";J;
      ")"
1990 INPUT X0(J,I),Z0(J,I)
2000 GOTO 1930
2010 J=-J ! DELETE CORNER
2020 IF J=N1(I) THEN 2080
2030 IF J>N1(I) THEN 2100
2040 FOR M=J TO N1(I)-1
2050 X0(M,I)=X0(M+1,I)
2060 Z0(M,I)=Z0(M+1,I)
2070 NEXT M
2080 N1(I)=N1(I)-1
2090 GOTO 1930
2100 NEXT I

```

```

2110 DISP "CHANGE SUSC? (Y/N)"
2120 INPUT Y0$
2130 IF Y0$#"Y" THEN 2180
2140 FOR I=1 TO N
2150 DISP "SUSC FOR BODY#";I
2160 INPUT K(I)
2170 NEXT I
2180 DISP "ADD BODS? (Y,N)"
2190 INPUT Y0$
2200 IF Y0$#"Y" THEN 1140
2210 N=N+1
2220 FOR I=N TO N
2230 GOTO 1040
2240 GRAPH ! ***DRAW PICTURE***
2250 PEN 1 @ GCLEAR
2260 H1=-99999
2270 FOR M=1 TO M0
2280 IF Z(M)>H1 THEN H1=Z(M)
2290 NEXT M
2300 X7=X9-X8
2310 Y8=H1-X7/4
2320 Y9=H1+X7/2
2330 SCALE X8,X9,Y8,Y9
2340 PEN 1
2350 MOVE X9,V2(V)
2360 DRAW X9,Y8
2370 DRAW X8,Y8
2380 DRAW X8,V2(1)
2390 FOR I=1 TO V
2400 DRAW V1(I),V2(I)
2410 NEXT I
2420 DRAW X9,V2(V)
2430 PENUP
2440 IF N<=0 THEN RETURN
2450 FOR M=2 TO M0 STEP 2
2460 PLOT X(M-1),Z(M-1)
2470 PLOT X(M),Z(M)
2480 PENUP
2490 NEXT M
2500 PENUP
2510 IF N<1 THEN RETURN
2520 FOR I=1 TO N
2530 N0=N1(I)
2540 MOVE X0(N0,I),Z0(N0,I)
2550 FOR J=1 TO N0
2560 DRAW X0(J,I),Z0(J,I)
2570 NEXT J
2580 NEXT I
2590 PAUSE
2600 ALPHA
2610 DISP "OKAY? (Y/N)"
2620 INPUT Y0$
2630 IF Y0$="N" THEN 1870
2640 RETURN
2650 ! ***PLOT FIELDS***
2660 H1=-99999
2670 L0=99999
2680 FOR M=1 TO M0
2690 IF F(M)<L0 THEN L0=F(M)
2700 IF F(M)>H1 THEN H1=F(M)

```

```

2710 NEXT M
2720 DISP "FMIN (<;L0;; FMAX >"
;H1
2730 INPUT Z8,Z9
2740 Z7=Z9-Z8
2750 Y8=Z9-1.5*Z7
2760 GRAPH @ PEN 1
2770 SCALE X8,X9,Y8,Z9
2780 XAXIS 0,D
2790 YAXIS X8,Z7/10,Z8,Z9
2800 Z5=(Z9-Y8)/(2*M0)
2810 FOR M=1 TO M0
2820 MOVE X(M),F(M)
2830 IMOVE D/4,Z5/2
2840 IDRAW -(D/2),-Z5
2850 IMOVE 0,Z5
2860 IDRAW D/2,-Z5
2870 NEXT M
2880 IF H(1)=0 THEN RETURN
2890 MOVE X(1),H(1)
2900 FOR M=1 TO M0
2910 DRAW X(M),H(M)
2920 NEXT M
2930 RETURN
2940 ! ***TAPE MAKER
2950 FOR I=1 TO N ! ADD Y1,Y2'S
2960 Y1(I)=-5000 ! TO FILL OUT
2970 Y2(I)=5000 ! TAPE FORMAT
2980 NEXT I
2990 DISP "INSERT DATA TAPE IN C
ARRIER"
3000 DISP "FILE NAME? (6 CHARS)"
3010 INPUT B$
3020 CREATE B$,23,240
3030 ASSIGN# 1 TO B$
3040 PRINT# 1 ; X8,X9,D,V3,M0,V,
N,T0,I0,S0
3050 PRINT# 1 ; N1(),K(),X(),Z()
,V1(),V2(),X0(),Z0(),H(),
Y1(),Y2(),F()
3060 ASSIGN# 1 TO *
3070 RETURN
3080 ! ***TAPE READER
3090 DISP "INSERT DATA TAPE IN C
ARRIER"
3100 DISP "FILE NAME? (6 CHARS)"
3110 INPUT B$
3120 ASSIGN# 1 TO B$
3130 READ# 1 ; X8,X9,D,V3,M0,V,N
,T0,I0,S0
3140 READ# 1 ; N1(),K(),X(),Z(),
V1(),V2(),X0(),Z0(),H(),Y
1(),Y2()
3150 ASSIGN# 1 TO *
3160 GOTO 1140

```

```

10 ! PROGRAM 2HDMAG***DAVE CAMP
    BELL, USGS-DENVER***SEPT 82
20 CLEAR
30 DISP "2HDMAG FORWARD CALC'N"
40 OPTION BASE 1
50 DEG
60 INTEGER I,J,M,M0,N,N0,N1(5),
    V
70 SHORT X0(20,5),Z0(20,5),X(10
    0),Z(100),F(100),V1(20),V2(2
    0),K(5),Y1(5),Y(5),H(100)
80 SHORT T0,I0,S0,X8,X9,D,V3,W,
    F0,X1,Z1,R1,P1,X2,Z2,R2,P2
90 REAL J1,J2,J3,P,S,C,U1,U2,W1
    ,G1,G2,G3,B1,B2,B3,F1,F2,F3,
    L,T2
100 SHORT Z8,Z9,Z7,H1,L0,Y8,Y9,X
    7,Z5 ! VARIABLES USED IN SUB
    S
110 IMAGE 2(DDDD.DDD,X),DDDDDD.DD
120 IMAGE 4X,2(DDDDDD.DD,X),DDDDDD
    .DD
130 H(1)=0
140 DISP "IS DATA ON TAPE? (Y/N)"
    "
150 INPUT Y0$
160 IF Y0$="Y" THEN 3370
170 DISP "TOTAL FIELD?"
180 INPUT T0
190 DISP "FIELD INCL (DEG)?"
200 INPUT I0
210 DISP "STRIKE ANGLE (DEG)?"
220 INPUT S0
230 DISP "***SET-UP FIELD POINTS
    ***"
240 DISP "FIELDX: XMIN,XMAX,DX"
250 INPUT X8,X9,D
260 M0=1+INT((X9-X8)/D)
270 FOR M=1 TO M0
280 X(M)=X8+(M-1)*D
290 NEXT M
300 DISP "INPUT OBSERVED FIELDS?
    (Y/N)"
310 INPUT Y0$
320 IF Y0$="Y" THEN 370
330 FOR M=1 TO M0
340 DISP "OBS(";X(M);")"
350 INPUT H(M)
360 NEXT M
370 DISP "INPUT TOPO? (Y/N)"
380 INPUT Y0$
390 IF Y0$="Y" THEN 680
400 ! ***TOPO INPUT**
410 DISP "HOW MANY TOPO POINTS"
420 INPUT V
430 FOR I=1 TO V
440 DISP "TOPO-X(";I;"),TOPO-Z("
    ;I;")"
450 INPUT V1(I),V2(I)
460 NEXT I

```

```

470 DISP "DRAPED SURVEY? (Y/N)"
480 INPUT Y0$
490 IF Y0$="Y" THEN 700
500 DISP "DRAPED DIST." ! **DRAP
    E
510 INPUT V3
520 FOR M=1 TO M0
530 W=X(M)
540 REM ***INTERPOLATION***
550 IF W>V1(1) THEN 580
560 F0=V2(1)
570 GOTO 650
580 IF W<V1(V) THEN 610
590 F0=V2(V)
600 GOTO 650
610 FOR J=2 TO V
620 IF W<V1(J) THEN 640
630 NEXT J
640 F0=V2(J-1)+(V2(J)-V2(J-1))*(
    W-V1(J-1))/(V1(J)-V1(J-1))
650 Z(M)=F0+V3
660 NEXT M
670 GOTO 770
680 V=1 ! ***CONST ALT SURVEYS
690 V1(1)=X8
700 DISP "FIELDZ"
710 INPUT E
720 FOR M=1 TO M0
730 Z(M)=E
740 NEXT M
750 IF V=1 THEN V2(1)=E
760 V3=0
770 ! ***BODY PARAMS**
780 DISP "HOW MANY BODIES?"
790 INPUT N
800 IF N>0 THEN 1070
810 DISP "TOPO SUSC?"
820 INPUT K(1)
830 DISP "TOPO HALF-LENGTH?"
840 INPUT Y(1)
850 L0=99999
860 FOR I=1 TO V
870 X0(I,1)=V1(I)
880 Z0(I,1)=V2(I)
890 IF V2(I)<L0 THEN L0=V2(I)
900 NEXT I
910 I=0
920 IF V2(V)=L0 THEN 980
930 X0(V+1,1)=2*X9-X8
940 Z0(V+1,1)=V2(V)
950 X0(V+2,1)=2*X9-X8
960 Z0(V+2,1)=L0
970 I=2
980 IF V2(1)=L0 THEN 1040
990 X0(V+I+1,1)=2*X8-X9
1000 Z0(V+I+1,1)=L0
1010 X0(V+I+2,1)=2*X8-X9
1020 Z0(V+I+2,1)=V2(1)
1030 I=I+2
1040 N=1

```

```

1050 N1(I)=V+I
1060 GOTO 1200
1070 FOR I=1 TO N
1080 DISP "FOR BODY NO.";I
1090 DISP "SUSC?"
1100 INPUT K(I)
1110 DISP "N CORNERS"
1120 INPUT N1(I)
1130 FOR J=1 TO N1(I)
1140 DISP "BOOX(";J;"),BOOZ(";J;")"
1150 INPUT X0(J,I),Z0(J,I)
1160 NEXT J
1170 DISP "HALF-LENGTH Y(";I;")"
1180 INPUT Y(I)
1190 NEXT I
1200 GOSUB 2540 ! **DRAW PICTURE
1210 DISP "PRINT BOOS? (Y/N)"
1220 INPUT Y0$
1230 IF Y0$#"Y" THEN 1360
1240 PRINT
1250 PRINT "FIELD=";T0;" INCL=";
      I0;" STRIKE=";S0
1260 FOR I=1 TO N
1270 PRINT
1280 PRINT "BODY #";I;"
      SUSC=";K(I)
1290 PRINT " HALF-LENGTH Y1=Y2="
      ;Y(I)
1300 PRINT " J      BOOX      BOOZ
      "
1310 FOR J=1 TO N1(I)
1320 IMAGE DD,2(X,DDDD.DDD)
1330 PRINT USING 1320 ; J;X0(J,I)
      ;Z0(J,I)
1340 NEXT J
1350 NEXT I
1360 RAD ! ***CALCULATE FIELD**
1370 FOR M=1 TO M0
1380 F(M)=0
1390 NEXT M
1400 FOR I=1 TO N
1410 N0=N1(I)
1420 J1=K(I)*T0*SIN(I0*PI/180)
1430 J2=K(I)*T0*COS(I0*PI/180)*S
      IN(S0*PI/180)
1440 J3=K(I)*T0*COS(I0*PI/180)*C
      OS(S0*PI/180)
1450 DISP
1460 DISP "***FOR BODY #";I
1470 DISP "      X      Z      Z
      FIELD"
1480 DISP "      XFIELD      YFIELD
      FIELD"
1490 FOR M=1 TO M0
1500 X1=X0(N0,I)-X(M)
1510 Z1=-Z0(N0,I)+Z(M)
1520 R1=SQR(X1^2+Z1^2)
1530 P1=SQR(R1^2+Y(I)^2)
1540 F1=0 @ F2=0 @ F3=0
1550 FOR J=1 TO N0
1560 X2=X0(J,I)-X(M)
1570 Z2=-Z0(J,I)+Z(M)
1580 R2=SQR(X2^2+Z2^2)
1590 P2=SQR(R2^2+Y(I)^2)
1600 P=ATN2(Z2-Z1,X2-X1)
1610 C=COS(P) @ S=SIN(P)
1620 U1=C*X1+S*Z1
1630 U2=C*X2+S*Z2
1640 W1=C*Z1-S*X1
1650 L=LOG(R1*(P2+Y(I))/(R2*(P1+
      Y(I))))
1660 T2=ATN2(U2*Y(I),W1*P2)-ATN2
      (U1*Y(I),W1*P1)
1670 IF T2<-PI THEN T2=T2+2*PI
1680 IF T2>PI THEN T2=T2-2*PI
1690 G1=C*J2+S*J1
1700 G2=C*J1-S*J2
1710 B1=-(C*G1*L)+C*G2*T2
1720 B2=S*G1*L-S*G2*T2
1730 B3=-(J3*T2)
1740 X1=X2
1750 Z1=Z2
1760 R1=R2
1770 P1=P2
1780 F1=F1+2*B1
1790 F2=F2+2*B2
1800 F3=F3+2*B3
1810 NEXT J
1820 F0=(J1*F1+J2*F2+J3*F3)/SQR(
      J1^2+J2^2+J3^2)
1830 F(M)=F(M)+F0
1840 DISP USING 110 ; X(M),Z(M),
      F1
1850 DISP USING 120 ; F2,F3,F(M)
1860 DISP
1870 NEXT M
1880 NEXT I
1890 DISP "PRINT OUTPUT? (Y/N)"
1900 INPUT Y0$
1910 IF Y0$#"Y" THEN 1980
1920 PRINT
1930 PRINT "SUM FIELD FOR ALL BO
      DIES"
1940 PRINT " FIELDX      FIELDZ
      FIELD"
1950 FOR M=1 TO M0
1960 PRINT USING 110 ; X(M),Z(M),
      F(M)
1970 NEXT M
1980 GOSUB 2950 ! ***PLOT FIELDS
1990 PAUSE
2000 ALPHA
2010 DISP "TAPE THIS CASE? (Y/N)"
      "
2020 INPUT Y0$
2030 IF Y0$#"Y" THEN 2050
2040 GOSUB 3240 ! **TAPE MAKER
2050 DISP "TRY CHANGES? (Y/N)"
2060 INPUT Y0$

```



```

2070 IF Y0$#"N" THEN 2100
2080 DISP "'BYE.....'"
2090 END
2100 DISP "CHANGE BODY CORNERS?
(Y/N)"
2110 INPUT Y0$
2120 IF Y0$#"Y" THEN 2340
2130 FOR I=1 TO N
2140 DISP "BODY #";I
2150 DISP " (TYPE 0 TO ESCAPE T
HIS BODY.)"
2160 DISP "CHANGE CORNER #"
2170 INPUT J
2180 IF J=0 THEN 2330
2190 IF J<0 THEN 2240
2200 IF J>N1(I) THEN N1(I)=J
2210 DISP "NEWX(";J);",NEWZ(";J;
")"
2220 INPUT X0(J,I),Z0(J,I)
2230 GOTO 2160
2240 J=-J
2250 IF J=N1(I) THEN 2310
2260 IF J>N1(I) THEN 2330
2270 FOR M=J TO N1(I)-1
2280 X0(M,I)=X0(M+1,I)
2290 Z0(M,I)=Z0(M+1,I)
2300 NEXT M
2310 N1(I)=N1(I)-1
2320 GOTO 2160
2330 NEXT I
2340 DISP "CHANGE SUSC? (Y/N)"
2350 INPUT Y0$
2360 IF Y0$#"Y" THEN 2410
2370 FOR I=1 TO N
2380 DISP "SUSC FOR BODY#";I
2390 INPUT K(I)
2400 NEXT I
2410 DISP "CHANGE STRIKE-LENGTHS
? (Y/N)"
2420 INPUT Y0$
2430 IF Y0$#"Y" THEN 2480
2440 FOR I=1 TO N
2450 DISP "Y FOR BODY#";I
2460 INPUT Y(I)
2470 NEXT I
2480 DISP "ADD BODS? (Y,N)"
2490 INPUT Y0$
2500 IF Y0$#"Y" THEN 1200
2510 N=N+1
2520 FOR I=N TO N
2530 GOTO 1080
2540 GRAPH ! **DRAW PICTURE**
2550 PEN 1 @ GCLEAR
2560 H1=-99999
2570 FOR M=1 TO M0
2580 IF Z(M)>H1 THEN H1=Z(M)
2590 NEXT M
2600 X7=X9-X8
2610 Y8=H1-X7/4
2620 Y9=H1+X7/2
2630 SCALE X8,X9,Y8,Y9
2640 PEN 1
2650 MOVE X9,V2(V)
2660 DRAW X9,Y8
2670 DRAW X8,Y8
2680 DRAW X8,V2(1)
2690 FOR I=1 TO V
2700 DRAW V1(I),V2(I)
2710 NEXT I
2720 DRAW X9,V2(V)
2730 PENUP
2740 IF N<=0 THEN RETURN
2750 FOR M=2 TO M0 STEP 2
2760 PLOT X(M-1),Z(M-1)
2770 PLOT X(M),Z(M)
2780 PENUP
2790 NEXT M
2800 PENUP
2810 IF N<1 THEN RETURN
2820 FOR I=1 TO N
2830 N0=N1(I)
2840 MOVE X0(N0,I),Z0(N0,I)
2850 FOR J=1 TO N0
2860 DRAW X0(J,I),Z0(J,I)
2870 NEXT J
2880 NEXT I
2890 PAUSE
2900 ALPHA
2910 DISP "OKAY? (Y/N)"
2920 INPUT Y0$
2930 IF Y0$="N" THEN 2100
2940 RETURN
2950 ! ***PLOT FIELDS***
2960 H1=-99999
2970 L0=99999
2980 FOR M=1 TO M0
2990 IF F(M)<L0 THEN L0=F(M)
3000 IF F(M)>H1 THEN H1=F(M)
3010 NEXT M
3020 DISP "FMIN (<);L0;"; FMAX >"
;H1
3030 INPUT Z8,Z9
3040 Z7=Z9-Z8
3050 Y8=Z9-1.5*Z7
3060 GRAPH @ PEN 1
3070 SCALE X8,X9,Y8,Z9
3080 XAXIS 0,0
3090 YAXIS X8,Z7/10,Z8,Z9
3100 Z5=(Z9-Y8)/(2*M0)
3110 FOR M=1 TO M0
3120 MOVE X(M),F(M)
3130 IMOVE 0/4,Z5/2
3140 IDRAW -(0/2),-Z5
3150 IMOVE 0,Z5
3160 IDRAW 0/2,-Z5
3170 NEXT M
3180 IF H(1)=0 THEN RETURN
3190 MOVE X(1),H(1)
3200 FOR M=1 TO M0
3210 DRAW X(M),H(M)

```

```

3220 NEXT M
3230 RETURN
3240 ! ***TAPE MAKER
3250 FOR I=1 TO N ! ADD Y1'S TO
3260 Y1(I)=-Y(I) ! FILL OUT
3270 NEXT I ! TAPE FORMAT.
3280 DISP "INSERT DATA TAPE IN C
ARRIER"
3290 DISP "FILE NAME? (6 CHARS)"
3300 INPUT B$
3310 CREATE B$,23,240
3320 ASSIGN# 1 TO B$
3330 PRINT# 1 ; ,X8,X9,D,V3,M0,V,
N,T0,I0,S0
3340 PRINT# 1 ; N1(),K(),X(),Z(),
,V1(),V2(),X0(),Z0(),H(),
Y1(),Y(),F()
3350 ASSIGN# 1 TO *
3360 RETURN
3370 ! ***TAPE READER
3380 DISP "INSERT DATA TAPE IN C
ARRIER"
3390 DISP "FILE NAME? (6 CHARS)"
3400 INPUT B$
3410 ASSIGN# 1 TO B$
3420 READ# 1 ; X8,X9,D,V3,M0,V,N
,T0,I0,S0
3430 READ# 1 ; N1(),K(),X(),Z(),
,V1(),V2(),X0(),Z0(),H(),Y
1(),Y()
3440 ASSIGN# 1 TO *
3450 GOTO 1200

```

NEWMAG 1/4

```

10  ! PROGRAM NEWMAG***SEPT,1982
    ****DAVE CAMPBELL, USGS-DENV
    ER
20  CLEAR
30  DISP "2HDMAG FORWARD CALC'N"
40  OPTION BASE 1
50  DEG
60  INTEGER I,J,M,M0,N,N0,N1(5),
    V
70  SHORT X0(20,5),Z0(20,5),X(10
    0),Z(100),F(100),V1(20),V2(2
    0),K(5),Y1(5),Y2(5),H(100)
80  SHORT T0,I0,S0,X8,X9,D,V3,W,
    F0,X1,Z1,R1,P1,Q1,X2,Z2,R2,P
    2,Q2
90  REAL J1,J2,J3,P,S,C,U1,U2,W1
    ,G1,G2,G3,B1,B2,B3,F1,F2,F3,
    L,T2,I4
100 SHORT Z8,Z9,Z7,H1,L0,Y8,Y9,X
    7,Z5 ! VARIABLES USED IN SUB
    S
110 IMAGE 2(DDDD.DDD,X),DDDDDD.DD
120 IMAGE 4X,2(DDDDDD.DD,X),DDDDDD
    .DD
130 H(1)=0
140 DISP "IS DATA ON TAPE? (Y/N)
    "
150 INPUT Y0$
160 IF Y0$="Y" THEN 3480
170 DISP "TOTAL FIELD?"
180 INPUT T0
190 DISP "FIELD INCL (DEG)?"
200 INPUT I0
210 DISP "STRIKE ANGLE (DEG)?"
220 INPUT S0
230 DISP "***SET-UP FIELD POINTS
    ***"
240 DISP "FIELDX: XMIN,XMAX,DX"
250 INPUT X8,X9,D
260 M0=1+INT((X9-X8)/D)
270 FOR M=1 TO M0
280 X(M)=X8+(M-1)*D
290 NEXT M
300 DISP "INPUT OBS FIELDS? (Y/N
    )"
310 INPUT Y0$
320 IF Y0$="Y" THEN 370
330 FOR M=1 TO M0
340 DISP "OBS( ";X(M);")"
350 INPUT H(M)
360 NEXT M
370 DISP "INPUT TOPO? (Y/N)"
380 INPUT Y0$
390 IF Y0$="Y" THEN 680
400 ! ***TOPO INPUT**
410 DISP "HOW MANY TOPO POINTS"
420 INPUT V
430 FOR I=1 TO V
440 DISP "TOPO-X( ";I;"),TOPO-Z(
    ";I;")"
450 INPUT V1(I),V2(I)
460 NEXT I
470 DISP "DRAPED SURVEY? (Y/N)"
480 INPUT Y0$
490 IF Y0$="Y" THEN 700
500 DISP "DRAPED DIST." ! **DRAP
    E
510 INPUT V3
520 FOR M=1 TO M0
530 W=X(M)
540 REM ***INTERPOLATION***
550 IF W>V1(1) THEN 580
560 F0=V2(1)
570 GOTO 650
580 IF W<V1(V) THEN 610
590 F0=V2(V)
600 GOTO 650
610 FOR J=2 TO V
620 IF W<V1(J) THEN 640
630 NEXT J
640 F0=V2(J-1)+(V2(J)-V2(J-1))*(
    W-V1(J-1))/(V1(J)-V1(J-1))
650 Z(M)=F0+V3
660 NEXT M
670 GOTO 770
680 V=1 ! ***CONST ALT SURVEYS
690 V1(1)=X8
700 DISP "FIELDZ"
710 INPUT E
720 FOR M=1 TO M0
730 Z(M)=E
740 NEXT M
750 IF V=1 THEN V2(1)=E
760 V3=0
770 ! ***BODY PARAMS**
780 DISP "HOW MANY BODIES?"
790 INPUT N
800 IF N>0 THEN 1070
810 DISP "TOPO SUSC?"
820 INPUT K(1)
830 DISP "TOPO END COORDS, Y1,Y2
    ?"
840 INPUT Y1(1),Y2(1)
850 L0=99999
860 FOR I=1 TO V
870 X0(I,1)=V1(I)
880 Z0(I,1)=V2(I)
890 IF V2(I)<L0 THEN L0=V2(I)
900 NEXT I
910 I=0
920 IF V2(V)=L0 THEN 980
930 X0(V+1,1)=2*X9-X8
940 Z0(V+1,1)=V2(V)
950 X0(V+2,1)=2*X9-X8
960 Z0(V+2,1)=L0
970 I=2
980 IF V2(1)=L0 THEN 1040
990 X0(V+I+1,1)=2*X8-X9
1000 Z0(V+I+1,1)=L0
1010 X0(V+I+2,1)=2*X8-X9

```

```

1020 Z0(V+I+2,1)=V2(1)
1030 I=I+2
1040 N=1
1050 N1(1)=V+I
1060 GOTO 1200
1070 FOR I=1 TO N
1080 DISP "FOR BODY NO.":I
1090 DISP "SUSC?"
1100 INPUT K(I)
1110 DISP "N CORNERS"
1120 INPUT N1(I)
1130 FOR J=1 TO N1(I)
1140 DISP "BODX(";J;"),BODZ(";J;
      ")"
1150 INPUT X0(J,I),Z0(J,I)
1160 NEXT J
1170 DISP "END COORDS Y1(";I;"),
      Y2(";I;")"
1180 INPUT Y1(I),Y2(I)
1190 NEXT I
1200 GOSUB 2680 ! **DRAW PICTURE
1210 DISP "PRINT BODS? (Y/N)"
1220 INPUT Y0$
1230 IF Y0$#"Y" THEN 1360
1240 PRINT
1250 PRINT "FIELD=";T0;" INCL=";
      I0;" STRIKE=";S0
1260 FOR I=1 TO N
1270 PRINT
1280 PRINT "BODY #";I;"      SUSC
      =" ;K(I)
1290 PRINT "Y1=";Y1(I);"      Y2=";
      Y2(I)
1300 PRINT " J      BODX      BODZ
      "
1310 FOR J=1 TO N1(I)
1320 IMAGE 00,2(X,0000.000)
1330 PRINT USING 1320 ; J;X0(J,I)
      ;Z0(J,I)
1340 NEXT J
1350 NEXT I
1360 RAD ! ***CALCULATE FIELD**
1370 FOR M=1 TO M0
1380 F(M)=0
1390 NEXT M
1400 FOR I=1 TO N
1410 N0=N1(I)
1420 J1=K(I)*T0*SIN(I0*PI/180)
1430 J2=K(I)*T0*COS(I0*PI/180)*S
      IN(S0*PI/180)
1440 J3=K(I)*T0*COS(I0*PI/180)*C
      OS(S0*PI/180)
1450 DISP
1460 DISP "***FOR BODY #":I
1470 DISP "      X      Z      Z
      FIELD"
1480 DISP "      XFIELD      YFIELD
      FIELD"
1490 FOR M=1 TO M0
1500 X1=X0(N0,I)-X(M)

```

```

1510 Z1=-Z0(N0,I)+Z(M)
1520 R1=SQR(X1^2+Z1^2)
1530 P1=SQR(R1^2+Y1(I)^2)
1540 Q1=SQR(R1^2+Y2(I)^2)
1550 F1=0 @ F2=0 @ F3=0
1560 FOR J=1 TO N0
1570 X2=X0(J,I)-X(M)
1580 Z2=-Z0(J,I)+Z(M)
1590 R2=SQR(X2^2+Z2^2)
1600 P2=SQR(R2^2+Y1(I)^2)
1610 Q2=SQR(R2^2+Y2(I)^2)
1620 P=ATN2(Z2-Z1,X2-X1)
1630 C=COS(P) @ S=SIN(P)
1640 U1=C*X1+S*Z1
1650 U2=C*X2+S*Z2
1660 W1=C*Z1-S*X1
1670 L=LOG(R1*(Q2+Y2(I))/(R2*(Q1
      +Y2(I))))
1680 L=L+LOG(R1*(P2-Y1(I))/(R2*(
      P1-Y1(I))))
1690 T2=ATN2(U2*Y2(I),W1*Q2)-ATN
      2(U1*Y2(I),W1*Q1)
1700 IF T2<-PI THEN T2=T2+2*PI
1710 IF T2>PI THEN T2=T2-2*PI
1720 T3=ATN2(U2*Y1(I),W1*P2)-ATN
      2(U1*Y1(I),W1*P1)
1730 IF T3<-PI THEN T3=T3+2*PI
1740 IF T3>PI THEN T3=T3-2*PI
1750 T2=T2-T3
1760 I4=0
1770 IF Y1(I)=0 THEN 1790
1780 I4=LOG((Q2-U2)*(Q1+U1)/((Q2
      +U2)*(Q1-U1)))/2
1790 IF Y2(I)=0 THEN 1810
1800 I4=I4-LOG((P2-U2)*(P1+U1)/(
      (P2+U2)*(P1-U1)))/2
1810 G1=C*J2+S*J1
1820 G2=C*J1-S*J2
1830 G3=G1*L-G2*T2-J3*I4
1840 B1=-(C*G3)
1850 B2=S*G3
1860 B3=(C*J1-S*J2)*I4-J3*T2
1870 X1=X2
1880 Z1=Z2
1890 R1=R2
1900 P1=P2
1910 Q1=Q2
1920 F1=F1+B1
1930 F2=F2+B2
1940 F3=F3+B3
1950 NEXT J
1960 F0=(J1*F1+J2*F2+J3*F3)/SQR(
      J1^2+J2^2+J3^2)
1970 F(M)=F(M)+F0
1980 DISP USING 110 ; X(M),Z(M),
      F1
1990 DISP USING 120 ; F2,F3,F0
2000 DISP
2010 NEXT M
2020 NEXT I

```

```

2030 DISP "PRINT OUTPUT? (Y/N)"
2040 INPUT Y0$
2050 IF Y0$#"Y" THEN 2120
2060 PRINT
2070 PRINT "SUM FIELD FOR ALL BO
DIES"
2080 PRINT " FIELDX FIELDZ
FIELD"
2090 FOR M=1 TO M0
2100 PRINT USING 110 ; X(M),Z(M)
,F(M)
2110 NEXT M
2120 GOSUB 3090 ! ***PLOT FIELDS
2130 PAUSE
2140 ALPHA
2150 DISP "TAPE THIS CASE? (Y/N)"
"
2160 INPUT Y0$
2170 IF Y0$#"Y" THEN 2190
2180 GOSUB 3380 ! **TAPE MAKER
2190 DISP "TRY CHANGES? (Y/N)"
2200 INPUT Y0$
2210 IF Y0$#"N" THEN 2240
2220 DISP "'BYE.....'"
2230 END
2240 DISP "CHANGE BODY CORNERS?
(Y/N)"
2250 INPUT Y0$
2260 IF Y0$#"Y" THEN 2480
2270 FOR I=1 TO N
2280 DISP "BODY #";I
2290 DISP " (TYPE 0 TO ESCAPE TH
IS BODY.)"
2300 DISP "CHANGE CORNER #"
2310 INPUT J
2320 IF J=0 THEN 2470
2330 IF J<0 THEN 2380
2340 IF J>N1(I) THEN N1(I)=J
2350 DISP "NEWX(";J;"),NEWZ(";J;
")"
2360 INPUT X0(J,I),Z0(J,I)
2370 GOTO 2380
2380 J=-J ! DELETE CORNER
2390 IF J=N1(I) THEN 2450
2400 IF J>N1(I) THEN 2470
2410 FOR M=J TO N1(I)-1
2420 X0(M,I)=X0(M+1,I)
2430 Z0(M,I)=Z0(M+1,I)
2440 NEXT M
2450 N1(I)=N1(I)-1
2460 GOTO 2380
2470 NEXT I
2480 DISP "CHANGE SUSC? (Y/N)"
2490 INPUT Y0$
2500 IF Y0$#"Y" THEN 2550
2510 FOR I=1 TO N
2520 DISP "SUSC FOR BODY#";I
2530 INPUT K(I)
2540 NEXT I
2550 DISP "CHANGE END COORDS Y1,
Y2? (Y/N)"

```

```

2560 INPUT Y0$
2570 IF Y0$#"Y" THEN 2620
2580 FOR I=1 TO N
2590 DISP "Y1,Y2 FOR BODY#";I
2600 INPUT Y1(I),Y2(I)
2610 NEXT I
2620 DISP "ADD BODS? (Y,N)"
2630 INPUT Y0$
2640 IF Y0$#"Y" THEN 1200
2650 N=N+1
2660 FOR I=N TO N
2670 GOTO 1080
2680 GRAPH ! ***DRAW PICTURE***
2690 PEN 1 @ GCLEAR
2700 H1=-99999
2710 FOR M=1 TO M0
2720 IF Z(M)>H1 THEN H1=Z(M)
2730 NEXT M
2740 X7=X9-X8
2750 Y8=H1-X7/4
2760 Y9=H1+X7/2
2770 SCALE X8,X9,Y8,Y9
2780 PEN 1
2790 MOVE X9,V2(V)
2800 DRAW X9,Y8
2810 DRAW X8,Y8
2820 DRAW X8,V2(1)
2830 FOR I=1 TO V
2840 DRAW V1(I),V2(I)
2850 NEXT I
2860 DRAW X9,V2(V)
2870 PENUP
2880 IF N<=0 THEN RETURN
2890 FOR M=2 TO M0 STEP 2
2900 PLOT X(M-1),Z(M-1)
2910 PLOT X(M),Z(M)
2920 PENUP
2930 NEXT M
2940 PENUP
2950 IF N<1 THEN RETURN
2960 FOR I=1 TO N
2970 N0=N1(I)
2980 MOVE X0(N0,I),Z0(N0,I)
2990 FOR J=1 TO N0
3000 DRAW X0(J,I),Z0(J,I)
3010 NEXT J
3020 NEXT I
3030 PAUSE
3040 ALPHA
3050 DISP "OKAY? (Y/N)"
3060 INPUT Y0$
3070 IF Y0$="N" THEN 2240
3080 RETURN
3090 ! ***PLOT FIELDS***
3100 H1=-99999
3110 L0=99999
3120 FOR M=1 TO M0
3130 IF F(M)<L0 THEN L0=F(M)
3140 IF F(M)>H1 THEN H1=F(M)
3150 NEXT M

```

```

3160 DISP "FMIN (<";L0;" FMAX >"
      ;H1
3170 INPUT Z8,Z9
3180 Z7=Z9-Z8
3190 Y8=Z9-1.5*Z7
3200 GRAPH @ PEN 1
3210 SCALE X8,X9,Y8,Z9
3220 XAXIS 0,0
3230 YAXIS X8,Z7/10,Z8,Z9
3240 Z5=(Z9-Y8)/(2*M0)
3250 FOR M=1 TO M0
3260 MOVE X(M),F(M)
3270 IMOVE D/4,Z5/2
3280 IDRAW -(D/2),-Z5
3290 IMOVE 0,Z5
3300 IDRAW D/2,-Z5
3310 NEXT M
3320 IF H(1)=0 THEN RETURN
3330 MOVE X(1),H(1)
3340 FOR M=1 TO M0
3350 DRAW X(M),H(M)
3360 NEXT M
3370 RETURN
3380 ! ***TAPE MAKER
3390 DISP "INSERT DATA TAPE IN C
      ARRIER"
3400 DISP "FILE NAME? (6 CHARS)"
3410 INPUT B$
3420 CREATE B$,23,240
3430 ASSIGN# 1 TO B$
3440 PRINT# 1 ; X8,X9,D,V3,M0,V,
      N,T0,I0,S0
3450 PRINT# 1 ; N1(),K(),X(),Z(),
      V1(),V2(),X0(),Z0(),H(),
      Y1(),Y2(),F()
3460 ASSIGN# 1 TO *
3470 RETURN
3480 ! ***TAPE READER
3490 DISP "INSERT DATA TAPE IN C
      ARRIER"
3500 DISP "FILE NAME? (6 CHARS)"
3510 INPUT B$
3520 ASSIGN# 1 TO B$
3530 READ# 1 ; X8,X9,D,V3,M0,V,N
      ,T0,I0,S0
3540 READ# 1 ; N1(),K(),X(),Z(),
      V1(),V2(),X0(),Z0(),H(),Y
      1(),Y2()
3550 ASSIGN# 1 TO *
3560 GOTO 1200

```

2D PLOT 1/2

CORRECTION FOR GRADIENTS

FOR MODELLING JOBS WHERE ONE ATTEMPTS TO MATCH AN OBSERVED FIELD, ONE MAY WISH TO ADD A LINEAR GRADIENT TO CALCULATED FIELD VALUES. TO DO SO, FIND THE FOLLOWING STEPS IN ANY OF THE PROGRAMS LISTED HERE

```
1270 RAD ! ***CALCULATE FIELD**
1280 FOR M=1 TO M0
1290 F(M)=0
1300 NEXT M
1310 FOR I=1 TO N
```

AND REPLACE THEM WITH THE FOLLOWING STEPS

```
1270 RAD ! ***CALCULATE FIELD**
1271 F(1)=0 @ D3=0
1272 DISP "REGIONAL GRADIENT? (Y/N)" @ INPUT Y0$
1273 IF Y0$#"Y" THEN 1280
1274 DISP "F(";X8;)" @ INPUT F(1)
1275 DISP "F(";X9;)" @ INPUT F0
1276 D3=F0-F(1) @ PRINT
1277 PRINT "REGIONAL HAS F(";X8;)"=";F(1);" AND F(";X9;)"=";F0
1280 FOR M=1 TO M0
1290 F(M)=F(1)+(M-1)*D3/(M0-1)
1300 NEXT M
1310 FOR I=1 TO N
```

THE VALUES YOU ENTER IN RESPONSE TO THE PROMPT, AND LINEARLY-INTERPOLATED VALUES IN BETWEEN, WILL THEN BE ADDED TO THE CALCULATED FIELD VALUES.

AUXILLARY PLOTTING PROGRAM

BECAUSE MANY USERS MAY WANT TO PLOT RESULTS FROM THESE PROGRAMS USING AUXILLARY PLOTTERS, A LISTING FOR PROGRAM "2D PLOT" IS GIVEN STARTING IN THE NEXT COLUMN. "2D PLOT" READS TAPES FROM THE OTHER PROGRAMS AND PLOTS THE "CRT PICTURE" TO SCALE ON THE IB-BUS PLOTTER. IT WARNS YOU IF THE SCALE YOU SPECIFY WON'T FIT THE PLOTTER PAPER. IT OUGHT TO WORK ON ALL HP PLOTTERS, BUT HAS ONLY BEEN CHECKED ON THE HP7470A AND HP9872S.

```
10 ! PROGRAM 2D PLOT***OCT, 1982*
***DAVE CAMPBELL, USGS-DENVER
20 CLEAR
30 DISP "PLOT 2HD PROFILES"
40 OPTION BASE 1
50 DEG
60 INTEGER I,J,M,M0,N,N0,N1(5),V
70 SHORT X0(20,5),Z0(20,5),X(100),Z(100),F(100),V1(20),V2(20),K(5),Y1(5),Y2(5),H(100)
80 SHORT T0,I0,S0,X8,X9,D,V3
90 SHORT Z8,Z9,Z7,H1,L0,Y8,Y9,X7,Z5 ! VARIABLES USED IN SUBS
100 IMAGE 2(DDDD.DDD,X),DDDD.DD
110 ! ***TAPE READER
120 DISP "INSERT DATA TAPE IN CARRIER"
130 DISP "FILE NAME? (6 CHARS)"
140 INPUT B$
150 ASSIGN# 1 TO B$
160 READ# 1 ; X8,X9,D,V3,M0,V,N,T0,I0,S0
170 READ# 1 ; N1(),K(),X(),Z(),V1(),V2(),X0(),Z0(),H(),Y1(),Y2(),F()
180 ASSIGN# 1 TO *
190 DISP "PRINT BODS? (Y/N)"
200 INPUT Y0$
210 IF Y0$#"Y" THEN 340
220 PRINT
230 PRINT "FIELD=";T0;" INCL=";I0;" STRIKE=";S0
240 FOR I=1 TO N
250 PRINT
260 PRINT "BODY #";I;" SUSC=";K(I)
270 PRINT "Y1=";Y1(I);" Y2=";Y2(I)
280 PRINT " J BODX BODZ"
290 FOR J=1 TO N1(I)
300 IMAGE DD,2(X,DDDD.DDD)
310 PRINT USING 300 ; J;X0(J,I);Z0(J,I)
320 NEXT J
330 NEXT I
340 DISP "PRINT OUTPUT? (Y/N)"
350 INPUT Y0$
360 IF Y0$#"Y" THEN 430
370 PRINT
380 PRINT "SUM FIELD FOR ALL BODIES"
390 PRINT " FIELDX FIELDZ FIELD"
400 FOR M=1 TO M0
410 PRINT USING 100 ; X(M),Z(M),F(M)
420 NEXT M
```

```

430 GRAPH ! ***DRAW PICTURE***
440 DISP "EXTERNAL PLOTTER? (Y/N)"
450 INPUT Y0$
460 IF Y0$#"Y" THEN 710
470 DISP "HOOK UP PLOTTER VIA IB BUS"
480 DISP "PLOTTER NUMBER? (e.g., 705)"
490 INPUT P0
500 PLOTTER IS P0
510 DISP "PAPER SIZE [mm,mm]?"
    (e.g., 190,250 FOR TYPING PAPER,"
520 DISP "270,370 FOR COMPUTER SHEETS.")
530 INPUT B1,B9
540 LIMIT 10,B9,10,B1
550 DISP "SCALE (e.g., 1/50000)"
560 INPUT S0
570 DISP "DISTANCES IN KFT OR KM?"
580 INPUT K0$
590 IF K0$="KM" THEN 620
600 IF K0$="KFT" THEN 620
610 GOTO 570
620 G9=10^8*(X9-X8)*S0/(B1-10)
630 IF K0$="KFT" THEN G9=G9/.3048
640 IF G9<133 THEN 670
650 DISP "WON'T FIT! TRY SCALE > 1/";133*S0/G9
660 GOTO 550
670 G0=0
680 IF G9<130 THEN G0=3
690 IF G9<120 THEN G0=10
700 LOCATE G0,G0+G9,G0,G0+3*G9/4
710 PEN 1 @ GCLEAR
720 H1=-99999
730 FOR M=1 TO M0
740 IF Z(M)>H1 THEN H1=Z(M)
750 NEXT M
760 X7=X9-X8
770 Y8=H1-X7/4
780 Y9=H1+X7/2
790 SCALE X8,X9,Y8,Y9
800 PEN 1
810 MOVE X9,V2(V)
820 DRAW X9,Y8
830 DRAW X8,Y8
840 DRAW X8,V2(1)
850 FOR I=1 TO V
860 DRAW V1(I),V2(I)
870 NEXT I
880 DRAW X9,V2(V)
890 PENUP
900 IF N<=0 THEN RETURN
910 FOR M=2 TO M0 STEP 2
920 PLOT X(M-1),Z(M-1)
930 PLOT X(M),Z(M)

```


```

940 PENUP
950 NEXT M
960 PENUP
970 IF N<1 THEN RETURN
980 FOR I=1 TO N
990 M0=M1(I)
1000 MOVE X0(N0,I),Z0(N0,I)
1010 FOR J=1 TO M0
1020 DRAW X0(J,I),Z0(J,I)
1030 NEXT J
1040 NEXT I
1050 PEN 0
1060 ALPHA
1070 H1=-99999
1080 L0=99999
1090 FOR M=1 TO M0
1100 IF F(M)<L0 THEN L0=F(M)
1110 IF F(M)>H1 THEN H1=F(M)
1120 NEXT M
1130 DISP "FMIN <;L0;; FMAX >"
    ;H1
1140 INPUT Z8,Z9
1150 Z7=Z9-Z8
1160 Y8=Z9-1.5*Z7
1170 GRAPH @ PEN 1
1180 SCALE X8,X9,Y8,Z9
1190 XAXIS 0,D
1200 YAXIS X8,Z7/10,Z8,Z9
1210 Z5=(Z9-Y8)/(2*M0)
1220 FOR M=1 TO M0
1230 MOVE X(M),F(M)
1240 IMOVE D/4,Z5/2
1250 IDRAW -(D/2),-Z5
1260 IMOVE 0,Z5
1270 IDRAW D/2,-Z5
1280 NEXT M
1290 IF H(1)=0 THEN 1240
1300 MOVE X(1),H(1)
1310 FOR M=1 TO M0
1320 DRAW X(M),H(M)
1330 NEXT M
1340 PEN 0
1350 ALPHA
1360 END

```


APPENDIX II: EXAMPLES

EXAMPLE: Rectangular body partially exhumed by sawtooth topography --



All distances in km.

Field points: X=0 to 10, step 0.25

	I	X	Z
Topo points(o):	1	2	0
	2	4	1
	3	5	0
	4	7	1
	5	8	0

Body points(x):	1	3	-1
	2	3	0.5
	3	4.5	0.5
	4	5	0
	5	6	0.5
	6	6.5	0.5
	7	6.5	-1

GRAVITY CALCULATIONS --- DENS = 0.1 g/cm^3 , DRAPED AT 0 KM

FOR 2HDGRV Y=1 FOR NEWGRV Y1=0.1, Y2=2.1

MAGNETICS CALCULATIONS --- SUSC= 0.001 cgs

T0 = 55000 nT IO = 60° SO = 50°

DRAPED AT 0.122 KM (=400 Ft)

FOR 2HD MAG Y=1 FOR NEWMAG Y1=0.1 Y2=2.1


2DGRAV KEYSTROKES

PROMPT	KEYSTROKE
	LOAD "2DGRAV"
	RUN
"IS DATA ON TAPE?"	N
"KFT OR KM?"	KM
"FIELDX: XMIN, XMAX, DX"	0, 10, .2 5
"INPUT OBS FIELDS?"	N
"INPUT TOPO?"	Y
"HOW MANY TOPO PTS?"	5
"TOPO-X(1), TOPO-Z(1)"	2, 0
2	2
3	3
4	4
5	5
"DRAPED SURVEY?"	Y
"Draped Dist"	0
"HOW MANY BODIES?"	1
"DENS"	0.1
"N CORNERS"	7
"BODX(1), BODZ(1)"	3, -1
2	2
3	3
4	4
5	5
6	6
7	7
PICTURE	CONT
"OKAY?"	Y
"PRINT BODS?"	Y
"PRINT OUTPUT?"	Y
"FMIN<.077457;	-0.5, 4.5
FMAX>4.3862"	
PICTURE	PAPER ADVANCE
	COPY
	CONT
"TAPE THIS CASE?"	N
"TRY CHANGES?"	N
"BYE....."	

2DGRAV RESULTS

BODY #	1	DENS= .1
J	BODX	BODZ
1	3.000	-1.000
2	3.000	.500
3	4.500	.500
4	5.000	0.000
5	6.000	.500
6	6.500	.500
7	6.500	-1.000

SUM FIELD	FOR ALL BODIES
FIELDX	FIELDZ
0.000	0.000
.250	0.000
.500	0.000
.750	0.000
1.000	0.000
1.250	0.000
1.500	0.000
1.750	0.000
2.000	0.000
2.250	.125
2.500	.250
2.750	.375
3.000	.500
3.250	.625
3.500	.750
3.750	.875
4.000	1.000
4.250	.750
4.500	.500
4.750	.250
5.000	0.000
5.250	.125
5.500	.250
5.750	.375
6.000	.500
6.250	.625
6.500	.750
6.750	.875
7.000	1.000
7.250	.750
7.500	.500
7.750	.250
8.000	0.000
8.250	0.000
8.500	0.000
8.750	0.000
9.000	0.000
9.250	0.000
9.500	0.000
9.750	0.000
10.000	0.000




BODY # 1 DENS= .1
 HALF-WIDTH= 1

J	BODX	BODZ
1	3.000	-1.000
2	3.000	.500
3	4.500	.500
4	5.000	0.000
5	6.000	.500
6	6.500	.500
7	6.500	-1.000

SUM FIELD FOR ALL BODIES

FIELDX	FIELDZ	FIELD
0.000	0.000	.02
.250	0.000	.03
.500	0.000	.03
.750	0.000	.04
1.000	0.000	.05
1.250	0.000	.06
1.500	0.000	.08
1.750	0.000	.11
2.000	0.000	.15
2.250	.125	.30
2.500	.250	.55
2.750	.375	.98
3.000	.500	1.90
3.250	.625	2.33
3.500	.750	2.34
3.750	.875	2.23
4.000	1.000	2.05
4.250	.750	2.62
4.500	.500	3.24
4.750	.250	2.85
5.000	0.000	2.30
5.250	.125	2.76
5.500	.250	2.92
5.750	.375	2.98
6.000	.500	2.94
6.250	.625	2.24
6.500	.750	1.52
6.750	.875	1.03
7.000	1.000	.74
7.250	.750	.55
7.500	.500	.36
7.750	.250	.20
8.000	0.000	.08
8.250	0.000	.06
8.500	0.000	.05
8.750	0.000	.04
9.000	0.000	.03
9.250	0.000	.03
9.500	0.000	.02
9.750	0.000	.02
10.000	0.000	.02




BODY # 1 DENS= .1
 Y1= 1 Y2= 2.1

J	BODX	BODZ
1	3.000	-1.000
2	3.000	.500
3	4.500	.500
4	5.000	0.000
5	6.000	.500
6	6.500	.500
7	6.500	-1.000

SUM FIELD FOR ALL BODIES

FIELDX	FIELDZ	FIELD
0.000	0.000	.02
.250	0.000	.02
.500	0.000	.03
.750	0.000	.03
1.000	0.000	.04
1.250	0.000	.05
1.500	0.000	.06
1.750	0.000	.08
2.000	0.000	.11
2.250	.125	.21
2.500	.250	.36
2.750	.375	.59
3.000	.500	.93
3.250	.625	1.21
3.500	.750	1.31
3.750	.875	1.32
4.000	1.000	1.28
4.250	.750	1.49
4.500	.500	1.55
4.750	.250	1.33
5.000	0.000	.97
5.250	.125	1.20
5.500	.250	1.31
5.750	.375	1.36
6.000	.500	1.34
6.250	.625	1.17
6.500	.750	.90
6.750	.875	.68
7.000	1.000	.53
7.250	.750	.40
7.500	.500	.26
7.750	.250	.15
8.000	0.000	.07
8.250	0.000	.05
8.500	0.000	.04
8.750	0.000	.03
9.000	0.000	.03
9.250	0.000	.02
9.500	0.000	.02
9.750	0.000	.02
10.000	0.000	.01



NEWGRV RESULTS

2HDGRV RESULTS

FIELD= 55000 INCL= 60
STRIKE= 50

BODY # 1 SUSC= .001
J BOOX BOOZ
1 3.000 -1.000
2 3.000 .500
3 4.500 .500
4 5.000 0.000
5 6.000 .500
6 6.500 .500
7 6.500 -1.000

SUM FIELD FOR ALL BODIES
FIELDX FIELDZ FIELD
0.000 .122 -12.60
.250 .122 -13.93
.500 .122 -15.49
.750 .122 -17.30
1.000 .122 -19.43
1.250 .122 -21.93
1.500 .122 -24.86
1.750 .122 -28.30
2.000 .122 -32.35
2.250 .247 -22.27
2.500 .372 1.73
2.750 .497 61.09
3.000 .622 211.84
3.250 .747 182.37
3.500 .872 145.47
3.750 .997 116.55
4.000 1.122 90.62
4.250 .872 83.58
4.500 .622 33.67
4.750 .372 -112.43
5.000 .122 -107.61
5.250 .247 26.98
5.500 .372 78.80
5.750 .497 111.42
6.000 .622 98.05
6.250 .747 -2.73
6.500 .872 -63.00
6.750 .997 -70.55
7.000 1.122 -62.72
7.250 .872 -76.52
7.500 .622 -74.57
7.750 .372 -63.13
8.000 .122 -49.16
8.250 .122 -40.46
8.500 .122 -33.88
8.750 .122 -28.80
9.000 .122 -24.79
9.250 .122 -21.57
9.500 .122 -18.94
9.750 .122 -16.77
10.000 .122 -14.96



FIELD= 55000 INCL= 60
STRIKE= 50

BODY # 1 SUSC= .001
J BOOX BOOZ
1 3.000 -1.000
2 3.000 .500
3 4.500 .500
4 5.000 0.000
5 6.000 .500
6 6.500 .500
7 6.500 -1.000

SUM FIELD FOR ALL BODIES
FIELDX FIELDZ FIELD
0.000 1.122 -2.20
.250 1.122 -1.50
.500 1.122 -.47
.750 1.122 1.01
1.000 1.122 3.16
1.250 1.122 6.26
1.500 1.122 10.73
1.750 1.122 17.19
2.000 1.122 26.52
2.250 1.122 39.88
2.500 1.122 58.42
2.750 1.122 81.62
3.000 1.122 103.82
3.250 1.122 115.60
3.500 1.122 114.61
3.750 1.122 105.21
4.000 1.122 90.62
4.250 1.122 71.76
4.500 1.122 50.70
4.750 1.122 34.11
5.000 1.122 27.09
5.250 1.122 27.47
5.500 1.122 29.53
5.750 1.122 26.88
6.000 1.122 13.60
6.250 1.122 -11.30
6.500 1.122 -39.03
6.750 1.122 -57.18
7.000 1.122 -62.72
7.250 1.122 -60.57
7.500 1.122 -55.32
7.750 1.122 -49.31
8.000 1.122 -43.49
8.250 1.122 -38.24
8.500 1.122 -33.64
8.750 1.122 -29.68
9.000 1.122 -26.28
9.250 1.122 -23.36
9.500 1.122 -20.86
9.750 1.122 -18.71
10.000 1.122 -16.85

2DMAG RESULTS--DRAPED

2DMAG RESULTS--LEVEL




FIELD= 55000 INCL= 60 STRIKE=
50

BODY # 1 SUSC= .001
HALF-LENGTH Y1=Y2= 1
J BODX BODZ
1 3.000 -1.000
2 3.000 .500
3 4.500 .500
4 5.000 0.000
5 6.000 .500
6 6.500 .500
7 6.500 -1.000

SUM FIELD FOR ALL BODIES

FIELDX	FIELDZ	FIELD
0.000	.122	-2.24
.250	.122	-2.59
.500	.122	-3.01
.750	.122	-3.52
1.000	.122	-4.12
1.250	.122	-4.83
1.500	.122	-5.65
1.750	.122	-6.55
2.000	.122	-7.51
2.250	.247	1.81
2.500	.372	23.99
2.750	.497	80.58
3.000	.622	228.01
3.250	.747	195.13
3.500	.872	155.14
3.750	.997	123.72
4.000	1.122	96.01
4.250	.872	106.41
4.500	.622	76.84
4.750	.372	-49.42
5.000	.122	-27.85
5.250	.247	100.38
5.500	.372	145.59
5.750	.497	171.96
6.000	.622	153.02
6.250	.747	47.32
6.500	.872	-17.51
6.750	.997	-29.45
7.000	1.122	-25.84
7.250	.872	-36.28
7.500	.622	-34.98
7.750	.372	-27.43
8.000	.122	-19.04
8.250	.122	-14.21
8.500	.122	-10.87
8.750	.122	-8.49
9.000	.122	-6.76
9.250	.122	-5.47
9.500	.122	-4.49
9.750	.122	-3.72
10.000	.122	-3.13

2HDMAG RESULTS



FIELD= 55000 INCL= 60 STRIKE= 50

BODY # 1 SUSC= .001
Y1= .1 Y2= 2.1
J BODYX BODYZ
1 3.000 -1.000
2 3.000 .500
3 4.500 .500
4 5.000 0.000
5 6.000 .500
6 6.500 .500
7 6.500 -1.000

SUM FIELD FOR ALL BODIES
FIELDX FIELDZ FIELD
0.000 .122 - .98
.250 .122 - .94
.500 .122 - .83
.750 .122 - .90
1.000 .122 - .61
1.250 .122 - .16
1.500 .122 .51
1.750 .122 2.14
2.000 .122 4.70
2.250 .247 16.25
2.500 .372 38.60
2.750 .497 81.88
3.000 .622 153.31
3.250 .747 152.35
3.500 .872 130.94
3.750 .997 109.61
4.000 1.122 89.16
4.250 .872 103.62
4.500 .622 99.75
4.750 .372 47.82
5.000 .122 41.25
5.250 .247 93.70
5.500 .372 116.66
5.750 .497 125.42
6.000 .622 108.21
6.250 .747 51.29
6.500 .872 6.54
6.750 .997 -9.25
7.000 1.122 -12.11
7.250 .872 -21.83
7.500 .622 -24.53
7.750 .372 -22.45
8.000 .122 -17.66
8.250 .122 -13.63
8.500 .122 -10.85
8.750 .122 -8.65
9.000 .122 -7.03
9.250 .122 -5.75
9.500 .122 -4.80
9.750 .122 -4.04
10.000 .122 -3.27

NEWMAG RESULTS--PROFILE S.E. OF BODY

FIELD= 55000 INCL= 60 STRIKE= 50

BODY # 1 SUSC= .001
Y1=-2.1 Y2=-.1
J BODYX BODYZ
1 3.000 -1.000
2 3.000 .500
3 4.500 .500
4 5.000 0.000
5 6.000 .500
6 6.500 .500
7 6.500 -1.000

SUM FIELD FOR ALL BODIES
FIELDX FIELDZ FIELD
0.000 .122 -3.31
.250 .122 -4.01
.500 .122 -4.91
.750 .122 -5.83
1.000 .122 -7.29
1.250 .122 -9.18
1.500 .122 -11.62
1.750 .122 -15.41
2.000 .122 -20.64
2.250 .247 -23.48
2.500 .372 -26.16
2.750 .497 -26.53
3.000 .622 -20.69
3.250 .747 -12.74
3.500 .872 -7.06
3.750 .997 -3.64
4.000 1.122 -2.11
4.250 .872 -28.49
4.500 .622 -87.96
4.750 .372 -129.09
5.000 .122 -169.86
5.250 .247 -142.39
5.500 .372 -124.06
5.750 .497 -106.49
6.000 .622 -80.53
6.250 .747 -59.28
6.500 .872 -47.90
6.750 .997 -36.29
7.000 1.122 -27.27
7.250 .872 -29.26
7.500 .622 -25.27
7.750 .372 -18.25
8.000 .122 -12.06
8.250 .122 -9.41
8.500 .122 -7.32
8.750 .122 -5.91
9.000 .122 -4.79
9.250 .122 -3.98
9.500 .122 -3.29
9.750 .122 -2.77
10.000 .122 -2.50

NEWMAG RESULTS--PROFILE N.W. OF BODY

